Beschreibung

ADODB („ActiveX Data Objects Databases“, meist, auch hier im weiteren Text, ADO genannt) ist der Nachfolger von DAO. Im Gegensatz zu DAO welches in Grundgerüst hauptsächlich für den Einsatz mit Jet-Datenbanken wie Access oder Excel  konzipiert ist und zu anderen Datenquellen meist über ODBC gehen muß stützt sich ADO auf die OLEDB-Treiber welche die ODBC-Treiber mehr und mehr ablösen. OLEDB bietet eine höhere Performance und oft auch mehr Möglichkeiten für den Entwickler. Über ADO lassen sich fast alle geläufigen Datenbanken zugreifen, egal ob Access, SQL-Server, Oracle, Informix und viele andere. Auch ODBC-Verbindungen, lassen sich problemlos verwenden, sind aber durch OLEDB eigentlich überflüssig.

Um ADO nutzen zu können muß im Menü unter Projekt-Verweise ein Verweis auf die „Microsoft ActiveX Data Objects 2.x Library“ gesetzt werden. Im weiteren Verlauf wird stellenweise ADO 2.5 vorausgesetzt.

Ein ADO-Datenmodell besteht im allgemeinen aus drei Hauptkomponenten:

Der Verbindung (Connection) welche die Kommunikation zur Datenbank erst ermöglicht. Sie enthält alle Grundlegenden Informationen über die Datenbank und dient als Basis für alle weiteren Aktionen. 

Zweitens dem Datensatz (Recordset), dieses Objekt gibt Daten in Form von Tabellen oder Auswahlabfragen (Views) zurück. Es dient sowohl zur Anzeige von Daten als auch zur Bearbeitung und Auswertung.

Das dritte Objekt ist das Befehls-Objekt (Command), hiermit kann man Aktionsabfragen (StoredProcedures) und SQL-Statements ausführen.

Vorwort

Wer vielleicht den Vorgänger dieses Tutorials kennt hat zwar bereits einen kleinen Grundstock zu ADO kennen gelernt, hier wird jedoch nun versucht auf den Umgang mit der ADO-Datenverarbeitung möglichst genau einzugehen und fast alle wichtigen und auch spezielleren Punkte genauer zu erläutern.

Hierzu erst einmal ein paar Vorinformationen welche sowohl wichtige Voraussetzungen als auch Punkte für die Weitergabe von ADO Datenbankprojekten beachtet werden müssen. Auf diese Punkte möchte ich zu Anfangs hier eingehen.

MDAC_TYP

Die MDAC_TYP („Microsoft Data Access Components“) sind die ADO-Treiber. Sie müssen auf einem Client vorhanden sein, damit dieser eine Anwendung die auf ADO basiert ausführen kann. Sie sollte eigentlich durch den Weitergabeassistenten mitgeliefert werden. Jedoch ist nicht sicher gestellt, daß auch die aktuelle Version mitgeliefert wird. Diese wird von Microsoft kostenlos zum Download angeboten unter www.microsoft.com/data/ . Bei Datenbanksystemen welche nicht von Microsoft stammen (z.B. Informix) werden die Treiber meist von jeweiligen Hersteller angeboten. Der Treiber für Oracle, welcher dem Oracle-Treiber von Microsoft klar vorzuziehen ist, ist in der MDAC von Microsoft enthalten.

Bei Windows ME und Windows 2000 ist die MDAC_TYP 2.1 ein Teil der Windowsinstallation, bei Windows XP ist ADO 2.7 Standart.

Beim Einsatz von Jet-Datenbanken (Access, dBase, FoxPro) und ODBC ist zu beachten, daß ADO 2.6 und (derzeit aktuellste) 2.7 diese Treiber nicht mehr beinhaltet. Hier müssen die Treiber entweder von einer Vorversion der MDAC_TYP bereits installiert sein (wird z.B. von MS-Office 2000 und MS-Office XP oder Windows 2000 und Windows XP mitgeliefert) oder extra installiert werden.

Die ODBC- und FoxPro-Treiber sind erhältlich unter:

http://msdn.microsoft.com/vfoxpro/downloads/odbc.asp
Die Jet-Treiber sind erhältlich unter:

http://support.microsoft.com/default.aspx?scid=kb;en-us;Q239114
DCOM_98

Bei Clients mit dem Betriebssystem Windows 95/98 muß, laut Microsoft, zusätzlich noch DCOM98 vorinstallliert werden. Nach meiner Sicht hat das zwar eher mit DirectX zu tun, aber man weiß ja nie… . DCOM98 ist auf der VisualBasic-/VisualStudio-CD im Verzeichnis Dcom98 zu finden (Dateiname „Dcom98.exe“).

Sicherheits-MDW

Im weiteren Verlauf werden bei Access oft Sicherheits-MDWs verwendet. Hierbei handelt es sich um MDB-ähnliche Dateien welche über den Arbeitsgruppenadministrator erstellt werden können. Sie ermöglichen in Verbindung mit Access eine recht gute Userverwaltung zu erstellen. Man sollte dringend vermeiden eigene Usertabellen anzulegen und dort Passwörter abzulegen, wer schon mal eine Access-MDB mit einem Editor geöffnet hat weiß warum! Auch eine Verschlüsslung der Datenbank durch Access halte ich für keine gute Lösung. Außerdem, wenn es diese Möglichkeit gibt sollte man sie doch nutzen. Zu erstellen setzen Sie sich am besten ein bißchen mit der Access-Hilfe auseinander.

Beispielumgebung

Zum leichteren Verständnis der nachfolgenden Beispiele hier eine kurze Beschreibung der für dieses Tutorial verwendeten Komponenten:

	Access2000

	Datenbankpfad
	"F:\Test_DBs\Access\db3.mdb"

	Datenbankpasswort
	"test"

	System-MDW-Pfad
	" F:\Test_DBs\Access\Secure.mdb "

	User-ID
	"florianr"

	User-Passwort
	"flo"

	Verwendeter Treiber
	"Microsoft.Jet.OLEDB.4.0"


	SQL-Server 2000

	Servername
	"FLO_MAIN"

	Datenbankname
	"db_Test"

	User-ID
	"florianr" (oder NT-Authentifikation)

	Passwort
	"flo" (oder NT-Authentifikation)


	ADODB

	Version
	2.5 (2.6 ist zwar bereits verfügbar wird jedoch von Microsoft noch nicht vollständig supported

	Standart-Connection-Name
	Cn oder CnT

	Standart-Recordset-Name
	Rs oder RsT

	Standart-Command-Name
	Cmd


SQL-Server und MSDE

Diesen Abschnitt sollten sich auf jeden sorgfältig Fall Access-Programmierer durchlesen!

Was ist der SQL-Server? Der SQL-Server ist das leistungsfähigste Pferd im Microsoftstall wenn es um Datenbanken geht.

Kleine Erklärung für reine Access-Programmierer:

Der SQL-Server ist nicht mit einer Access-Datenbank vergleichbar. Auf einem Server werden mehrere Datenbanken verwaltet. Der Server kann entweder über einen zugeteilten NT (2000) Usernamen erreicht werden oder auf eine eigene, interne, Sicherheit zurückgreifen. Der SQL-Server ist wesentlich leistungsfähiger als Access und sicherer. Es ist möglich über StoredProcedures und Trigger zu erstellen, hierbei handelt es sich um leistungsstarke und z.T. große SQL-Skripts welche die Aktionsabfragen von Access weit in den Schatten stellen. Die drei wichtigsten Komponenten des SQL-Servers sind der Enterprise-Manager, der Service-Manager und der Query-Analyser, diese Komponenten werden weiter unten noch genauer beschrieben.

Es gibt viele verschiedene Versionen des SQL-Servers hier nur ein paar davon.

Enterprise Edition

Unterstützt bis zu 32 CPUs unterstützt (sowie Clustering), einen Arbeitsspeicher von bis 2 GByte und eine unbegrenzte Größe der Datenbanken. Als Plattform wird NT oder Windows 2000 Enterprise Server vorausgesetzt. Er ist nur für riesige Datenbanksysteme im Bereich von mehreren GByte oder TByte sinnvoll.

Standart Edition

Unterstützt bis zu 4 CPUs, 2 GBytes RAM. Unlimitierte Datenbankgrößen. Plattform muß ein NT- oder 2000- Server sein.

Developer Edition

Ähnlich der Standart-Edition. Es dürfen jedoch nur Tests darauf ausgeführt werden. Die Anzahl der möglichen Clients ist auf 10 begrenzt. Kostenloser Bestandteil von VisualStudio- und VisualBasic-Enterpriseedition.

MSDE

Was ist die MSDE (Microsoft Data Engine)? Die MSDE ist eine geniale Neuerung von Microsoft, hierbei handelt es sich um eine abgespeckte Variante der SQL-Server. Die kann bis zu 2 GByte RAM verwalten und erlaubt eine Datenbankgröße von bis zu 2 GByte, sie kann auf normalen Windows-Clients (wie 9x, NT, 2000, …) installiert werden. Die MSDE bietet im eigentlichen Lieferumfang keine Entwicklungsumgebungen wie den Enterprise-Manager und den Query-Analyser, sondern nur den Service-Manager.

Laut Microsoft kann die MSDE immer nur 5 User gleichzeitig bedienen. Dies bedeutet jedoch nicht, daß nur 5 User gleichzeitig eingeloggt sein können sondern, daß nur 5 Aktionen (z.B. StoredProcedures) gleichzeitig ausgeführt werden können. Test mit 25 Aktionen gleichzeitig ergaben keinerlei Zeitunterschied zum SQL-Server.

Wichtig ist auch, daß die MSDE nicht gleichzeitig mit dem SQL-Server installiert sein darf! Beide Systeme verwenden die gleichen Verzeichnisse und Registry-Einträge. Leider wird jedoch bei der Installation keine Fehlermeldung ausgegeben sondern einfach alte Daten überschrieben. Natürlich werden aber nur die Programmdateien, nicht aber die Datenbanken überschrieben.

SQL-Server vs. MSDE

Nach dem bisherigen Text könnte man sagen, die MSDE ist ja einfach nur schwächer und unkomfortabler als der SQL-Server. Der einzige Vorteil könnte sein, daß Sie keinen NT-/2000-Server als Plattform benötigen.

Die MSDE hat jedoch ganz gravierende Vorteile! Alle anderen Versionen (bis auf die Developer-Edition) kosten viel Geld. Die MSDE ist kostenlos (!) und kann auch lizenzfrei mit Datenbankprogrammen beliebig vertrieben werden.

Die MSDE ist auf verschiedenen Wegen erhältlich, sie ist beispielsweise im Lieferumfang von Microsoft Office 2000 Premium oder Developer enthalten (Office-CD 1, Pfad CD-ROM\SQL\X86\Setup\SetupSQL.exe). Sie kann von lizenzierten Besitzern von VisualBasic (Pro/Enterprise) oder VisualStudio (Pro/Enterprise) bei Microsoft bestellt werden, im Lieferumfang ist dann auch der Enterprisemanager Developer-Edition und der Query-Analyser welche beide für die Entwicklung von zu vertreibender Software durchaus genutzt werden können, nur nicht im Live-Betrieb verwendet werden dürfen. Außerdem sind beide Komponenten im Lieferumfang des VisualStudios Enterprise-Edition enthalten (aber Version 6.5 und nicht 7, also 2000) unter CD-ROM\SQL\I386\SETUP.EXE . Wichtig ist jedoch, daß hierbei nicht die Developer-Edition des SQL-Servers sondern nur der Enterprise-Manager und der Query-Analyser installiert werden! Außerdem kann die MSDE auch durchaus gut mit Access 2000 verwaltet werden. Access 2000 bietet eine direkte Schnittstelle zum SQL-Server und eine sehr schöne Entwicklungsumgebung (unverständlicherweise auch eine wesentlich größere als für Accesseigene Datenbanken!).

Außerdem sollte eine maximale Datenbankgröße (nicht gesamter Server!) von 2 GByte doch eine Zeit durchaus reichen.

Der Gedanke von Microsoft war, daß bislang Access die einzige kostenlose Datenbankplattform war welche kostenfrei veräußert werden konnte. Durch die eher geringe Leistungsfähigkeit und Fremdheit zu richtigen DB-Systemen war kein Grund sich später für ein anderes Microsoftprodukt (SQL-Server zu entscheiden). Durch die MSDE bietet Microsoft nun eine sehr leistungsfähige Grundlage. Wenn man auf der MSDE angefangen hat und nun die MSDE doch irgendwann nicht mehr ausreichen sollte ist jedoch der Schritt auf den SQL-Server sehr klein, da die MSDE 1:1 mit dem SQL-Server kompatibel ist.

Service-Manager

Das ist eigentlich nur ein sehr kleines Tool welches nichts mit der Entwicklung von Datenbanken zu tun hat. Hierbei handelt es sich um eine kleine Konsole welche die Dienste für die MSDE (den SQL-Server) starten, anhalten und beenden kann. 

Wenn die Dienste jedoch nicht laufen kann die Datenbank nicht angesprochen werden.

Enterprise-Manager

Der Enterprise-Manager ist eigentlich das Herz des SQL-Servers. Hiermit lassen sich neue User/Logins/Datenbanken/… anlegen und verwalten. Der Enterprise-Manager gibt eine Übersicht über alle Server/Datenbanken und deren Objekte wie StoredProcedures/Views/… . Er bietet einen kleinen SQL-Editor (leider nur für Views).

Query-Analyser

Dieses Tool hat zwar keinen SQL-Assistenten wie Access und der Enterprise-Manager, er ist jedoch ein sehr wichtiges Tool für den Umgang mit dem SQL-Server. Er bietet die Möglichkeit Datenbankabfragen zu testen, StoredProcedures, Views und andere SQL-Server spezifischen Objekte anzulegen. Außerdem bietet der Query-Analyser die Möglichkeit die SQL-Statements zu debuggen und Analysen über Performance und vieles anderes auszugeben.

Properties

Bevor wir auf die eigentlichen Objekte und deren Funktionalität von ADO eingehen wäre die Properties-Collection, welche sich bei fast allen ADO-Objekten wiederfindet zu erwähnen. Der Gedanke diese zur Verfügung zu stellen war eine sehr gute Idee von Microsoft! Hierbei handelt es sich um eine Möglichkeit der verschiedenen Datenbankprovider eigene Eigenschaften zur Verfügung zu stellen. Eine einzelne Property besitzt vier Eigenschaften. 

Die Eigenschaft Attributes spielt jedoch eine sehr untergeordnete Rolle und ist noch lange nicht für alle Provider gleichermaßen verfügbar. Falls sie bei einem Objekt von Wichtigkeit ist wird sie weiter unten explizit erwähnt. 

Die Eigenschaft Name stellt einen String dar welcher den Namen der Property angibt, da es sich bei den Properties um eine Collection handelt kann der Name auch als Indikator genutzt werden

Die Properties können über eine sehr kleine Routine erfragt werden:

Private Sub Command1_Click()

Dim i As Integer

Dim Cn As ADODB.Connection

Set Cn = New ADODB.Connection

Cn.Provider = "SQLOLEDB.1"

For i = 0 To Cn.Properties.Count - 1

    Debug.Print Cn.Properties(i).Name,

    Debug.Print Cn.Properties(i).Value

Next i

End Sub

Verbindung (Connection)

Wie bereits erwähnt ist die Connection der grundliegende Baustein zwischen der Datenbank und dem Frontend (der Anwendung). Eine Connection kann mehrere untergeordnete Objekte wie Recordsets und Commands gleichzeitig handeln, somit reicht es normalen Anwendungen meist eine einzige Connection für das gesamte Projekt zu verwenden (diese kann z.B. global in einem Modul gehalten werden).

Die Connection bietet zwar im Grundmodell nur verhältnismäßig wenige Eigenschaften und Methoden, diese sind jedoch durch die Basisfunktionalität der Connection um so stärker in ihrer Kraft.

Provider

Der Provider gibt der Connection an was für eine Art von Datenquelle angesteuert werden soll und welche Version der Datenquelle vorliegt. Der Provider wird über einen String definiert.

Wichtige Provider:

	Datenbanktreiber

	Datenbank
	Provider

	MS-SQL-Server (OLEDB)
	"SQLOLEDB.1"

	MS-SQL-Server (ODBC)
	"MSDASQL.1"

	MS-Access97
	"Microsoft.Jet.OLEDB.3.51"

	MS-Access2000
	"Microsoft.Jet.OLEDB.4.0"

	Oracle-DB (von Oracle)
	"OraOLEDB.Oracle.1"

	Oracle-DB (von Microsoft)
	"MSDAORA.1"


ACHTUNG für Access-Benutzer

Sobald der Access-Treiber 4.0 installiert ist wird der 3.5-Treiber deaktiviert und auch für Access97-Datenbanken der 4.0-Treiber verwendet. Dies sollte auch im Code so gehandelt werden, da es sonst eventuell zu Problemen beim ausliefern des Programms an den Kunden führen kann.
Beispiel für SQL-Server:

Option Explicit

Dim Cn As New ADODB.Connection

Private Sub Form_Load()

Cn.Provider = "SQLOLEDB.1"

End Sub

Beispiel Access:

Option Explicit

Dim Cn As New ADODB.Connection

Private Sub Form_Load()

Cn.Provider = "Microsoft.Jet.OLEDB.4.0"

End Sub
ConnectionString

Der ConnectionString kann genutzt werden um alle weiteren Informationen der Datenbankverbindung durch eine Aneinanderreihung von Eigenschaften getrennt mit einem „;“ zu übergeben. Im allgemeinen ist der ConnectionString jedoch eher unübersichtlich und führt leicht zu Fehlern:

Für Access2000

Option Explicit

Dim Cn As New ADODB.Connection

Private Sub Form_Load()

Cn.Provider = "Microsoft.Jet.OLEDB.4.0"

Cn.ConnectionString = "Data Source=C:\Temp\Test.mdb"

End Sub

Für SQL-Server:

Private Sub Form_Load()

With Cn

    ' Provider zuweisen

    .Provider = "SQLOLEDB.1"

    ' ConnectionString übergeben

    .ConnectionString = "Data Source=FLO_MAIN;" & _

                        "User Id=florianr;" & _

                        "Password=flo"

    ' Connection öffnen

    .Open

End With

End Sub

Properties der Connection

Eine vielleicht etwas ausführlichere aber dafür wesentlich übersichtlicher und auch leichter veränderbare Methode zur Übergabe der Connection-Informationen liegt darin die Properties-Collection zu nutzen. Alle Parameter des ConnectionStrings sind ebenfalls in dieser vorhanden.

Aufruf ist:

Cn.Properties("PropertyName") = "PropertyWert"

Hier ein paar der wichtigsten Properties mit Erläuterung. Es sind nur ein Bruchteil der Properties (beim SQL-Server sind es 93 Properties!):

Data Source

Diese Property findet man eigentlich bei fast allen DB-Treibern wieder. Somit hat sie auch dementsprechend verschiedene Hintergründe. Bei Access beschreibt der DataSource den Pfad zu der physischen MDB-Datei. Beim MS-SQL-Server beschreibt der DataSource den Namen des Servers (nicht die Datenbank!). Bei Informix z.B. beschreibt diese Eigenschaft wieder den Server und den Namen der Datenbank. Hier zwei Beispiele:

Für Access:

Cn.Properties("Data Source") = "F:\Test_DBs\Access\db3.mdb"
Für SQL-Server:

Cn.Properties("Data Source") = "FLO_MAIN"

Initial Catalog

Diese Eigenschaft wird vom OLEDB-Treiber des SQL-Servers. Diese Eigenschaft gibt den Namen der Datenbank auf welche zugegriffen werden soll an.

Integrated Security

Ebenfalls eine Eigenschaft des SQL-Servers. Sie gibt an ob eine User-ID und ein Passwort von Client übergeben werden müssen oder ob der SQL-Server auf die NT-Authentifikation zurückgreift. Soll von ID und Passwort abgesehen werden muß hier „SSPI“ übergeben werden:

Option Explicit

Dim Cn As ADODB.Connection

Private Sub Form_Load()

Set Cn = New ADODB.Connection

Cn.Provider = "SQLOLEDB.1"

Cn.Properties("Data Source") = "FLO_Main"

Cn.Properties("Initial Catalog") = "db_Test"

Cn.Properties("Integrated Security") = "SSPI"

Cn.Open

End Sub

Jet OLEDB:Database Password

Datenbankpasswort einer Accessdatenbank. Dieses darf nicht mit dem Userpasswort aus den Logininformationen einer MDW-Sicherheitsdatei verwechselt werden.

Jet OLEDB:System database

Pfad zu der System-MDW (Sicherheitdatei).

User ID

User-ID des anzumeldenden Users.

Password

Passwort der anzumeldenden Users.

Persist Security Info

Diese Eigenschaft gibt an, ob nach dem Aufbau der Connection zur Datenbank die Userinformationen aus dem ConnectionString und den Connection-Properties gelöscht werden sollen oder gehalten werden sollen. Die Default-Einstellung ist unverständlicher Weise True (behalten). Dieser Wert sollte also explizit auf False gesetzt werden.

Prompt

Prompt ist eine sehr feine Sache die ADO bietet. Hierbei spart man sich das erstellen eines Anmeldeformulars für den Login in der Datenbank. Prompt stellt, je nach Einstellung ein Standartanmeldefenster für die Datenbank dar. Je nach Einstellung kann entschieden werden ob das Fenster immer gezeigt wird oder nur wenn Anmeldedaten fehlen. Auch die Menge der einzugebenden Daten kann angegeben werden, wobei die ausführliche Anzeige eigentlich nur für Datenbankreporter oder ähnliche Software sinnvoll ist, da hier zusätzlich zum Servernamen und den Logindaten auch noch die Datenbank, die Sprache u. a. gewählt werden kann. Man kann z.B. über SaveSettings/GetSettings die UserId speichern und beim nächsten Programmstart wieder vor der Anzeige setzen, so daß der User nur noch sein Passwort eingeben muß.

Beim Einsatz mit Jet-DBs (Access) ist diese Eigenschaft etwas mit Vorsicht zu genießen, weil der User hier den Mode (siehe nächster Punkt) angeben kann und somit evtl. anders auf die DB zugreifen kann als es eigentlich gewünscht ist. Dies kann jedoch durch entsprechende Einstellungen in der System-MDW verhindert werden.

Die verschiedenen Möglichkeiten für die Prompts werde durch das ConnectPromptEnum definiert.

	ConnectPromptEnum

	Konstante
	Beschreibung

	adPromptAlways
	Das Fenster wird auf jeden Fall gezeigt (auch wenn die Anmeldedaten eigentlich komplett sein könnten).

	adPromptComplete
	Das Fenster wird angezeigt wenn nicht alle Daten vorhanden sind. Es sind alle Optionen möglich.

	adPromptCompleteRequired
	Das Fenster wird gezeigt wenn nicht alle Daten vorhanden sind. Es können nur User-Id, Passwort und Servername (SQL-Server) angegeben werden.

	adPromptNever
	Erlaubt im MultiUser-Bereich das öffnen anderer Verbindungen ohne Beschränkungen (Lesen/Schreiben/Beides).


Beispiel für Access:

Private Sub Form_Load()

Dim promp As ConnectPromptEnum

Dim sUserId As String

' Prompt soll zur Verfollständigung aber nur mit

' Möglichkeiten für normalen User angezeigt werden

promp = adPromptCompleteRequired

' Bestehenden Wert aus der Registry auslesen,

' wenn kein Wert vorhanden dann NT-UserId auslesen

sUserId = GetSetting(App.Title, "Main", "UserId", _

                        Environ$("username"))

With CnT

    .CursorLocation = adUseClient

    .Mode = adModeShareDenyNone

    .Provider = "Microsoft.Jet.OLEDB.4.0"

    .Properties("Data Source") = "F:\Test_DBs\Access\db3.mdb"

    .Properties("Jet OLEDB:Database Password") = "test"

    .Properties("Jet OLEDB:System database") = _

                            "F:\Test_DBs\Access\db3.mdw"

    .Properties("User ID") = sUserId ' UserId übergeben

    .Properties("Prompt") = promp   ' Prompt-Mode übergeben

    .Open

    sUserId = .Properties("User Id").Value

End With

' UserId für nächsten Programmstart speichern

SaveSetting App.Title, "Main", "UserId", sUserId

End Sub

Beispiel für SQL-Server

Private Sub Form_Load()

Dim promp As ConnectPromptEnum

Dim sUserId As String

' Prompt soll zur Verfollständigung angezeigt werden.

' Es können mehrere Einstellungen vorgenommen werden.

promp = adPromptComplete

' Bestehenden Wert aus der Registry auslesen,

' wenn kein Wert vorhanden dann NT-UserId auslesen

sUserId = GetSetting(App.Title, "Main", "UserId", _

                        Environ$("username"))

With CnT

    .Provider = "SQLOLEDB.1"

    .Properties("Data Source") = "FLO_MAIN"

    .Properties("Initial Catalog") = "db_Test"

    .Properties("User ID") = sUserId

    .Properties("Prompt") = promp

    .Open

End With

' UserId für nächsten Programmstart speichern

SaveSetting App.Title, "Main", "UserId", sUserId

End Sub

Beispiel für Properties

Um mal ein paar der beschriebenen Properties einzusetzen hier die Mammut-Routine um eine Access-Datenbank mit Datenbankpasswort und Sicherheitsdatei zu öffnen:

With Cn

    .Provider = "Microsoft.Jet.OLEDB.4.0"

    .Properties("Data Source") = "F:\Test_DBs\Access\db3.mdb"

    .Properties("Jet OLEDB:Database Password") = "test"

    .Properties("Jet OLEDB:System database") = _ 

                                  "F:\Test_DBs\Access\db3.mdw"

    .Properties("User ID") = "florianr"

    .Properties("Password") = "flo"

    .Open

End With

Mode

Diese Eigenschaft ist mir bislang nur von Jet-Datenbanken geläufig. Sie gibt prinzipiell an wie der Zugriff der Anwendung auf die DB ist und wie sich die DB im Multiuserbetrieb verhält. Beim Einsatz dieser Eigenschaft beim SQL-Server wird ein Fehler ausgelöst.

	ConnectModeEnum

	Konstante
	Beschreibung

	adModeRead
	Nur Lesen.

	adModeReadWrite
	Lesen und schreiben

	adModeRecursive
	Wird verwendet in Kombination mit anderen MultiUser-Eigenschaften (ShareDeny*) die Beschränkungen werden dann auch auf Sub-Records (abgeleitete Datensätze) übergeben. Wenn keine Sub/Child-Records existieren besteht keine Funktionalität.

	adModeShareDenyNone
	Erlaubt im MultiUser-Bereich das öffnen anderer Verbindungen ohne Beschränkungen (Lesen/Schreiben/Beides).

	adModeShareDenyRead
	Verbietet anderen Connections einen Lesezugriff.

	adModeShareDenyWrite
	Verbietet anderen Connections einen Schreibzugriff.

	adModeShareExclusive
	Verbietet das öffnen anderer Connections grundsätzlich.

	adModeUnknown
	Keine Beschrenkungen (= Default).

	adModeWrite
	Nur Schreiben.


Execute

Diese Methode kann genutzt werden um SQL-Statements abzuschicken. Rückgabewert ist ein ReadOnly-Recordset (wird weiter unten noch beschrieben).

Beispiel zum abschicken eines SQL-Statements:

Cn.Execute "INSERT INTO tblName (FeldName) VALUES ('Hallo')"

BeginTrans, RollbackTrans, CommitTrans

Diese Methoden sollten denjenigen die bereits den Umgang mit dem SQL-Server kennen bereits bekannt vorkommen.

Sie sind vor allem im Umgang mit Errorhandling sehr praktisch um eventuell halbe Einträge in Datenbanken zu vermeiden.

BeginTrans dient zum Festlegen des Beginns einer SQL-Transaktion. Mit CommitTrans wird die Transaktion vollendet, wenn jedoch während der Transaktion ein Fehler auftritt können mit der Methode RollbackTrans die bereits durchgeführten Transaktionen rückgängig gemacht werden. Hier ein kleines Beispiel:

Private Sub Command1_Click()

On Error GoTo ErrSQL

Cn.BeginTrans

Cn.Execute "DELETE FROM tblName WHERE Spalte1 = 'Hallo'"

Cn.CommitTrans

ErrSQL:

Cn.RollbackTrans

End Sub

CommandTimeout

Gibt an innerhalb welcher Zeit in Sekunden eine Transaktion ausgeführt sein muß bevor diese von der Connection abgebrochen wird. Default ist 30 Sekunden.

CursorLocation

Gibt an wo die Daten gehalten werden sollen wenn man eine Datenabfrage auf die Datenbank ausgeführt wird. Es gibt hier zwei Möglichkeiten:

	CursorTypeEnum

	Konstante
	Beschreibung

	adUseClient
	Bedeutet, daß die Daten komplett auf den Client gezogen werden sollen. Dies bringt jedoch einen eventuell großen Speicherbedarf auf dem Client.

	adUseServer
	Die Daten werden nicht komplett auf den Client gezogen, sondern nur diese die vom Client aktuell benötigt werden. Das spart viel Speicherbedarf auf dem Client. Nachteil ist jedoch, daß bei manchen Datenanzeige-Controlls wie z.B. DataGrid oder MSHFlexGrid die Daten mit einem serverseitigen Cursor keine Daten angezeigt werden, bzw. ein paar Funktionen nicht mehr verfügbar sind.


Errors

Im Gegensatz zu VB können bei ADO und SQL durch eine einzige Methode mehrere Fehler und Meldungen vom Server an den Client geschickt werden. Aus diesem Grund ist Errors bei ADO kein einzelnes Objekt sondern eine Connetion.

State

Gibt den aktuellen Status der Connection an. Die wichtigsten für den normalen Umgang mit ADO sind adStateOpen für Connection ist offen und adStateClosed für Connection ist geschlossen.

Cancel

Diese Methode ist für das abbrechen einer asynchronen Abfrage oder SQL-Action gedacht. Hierauf wird aber erst später eingegangen.

OpenSchema

Hierbei handelt es sich um eine sehr schöne Methode ohne den Einsatz von zusätzlichen Datenbankverwaltungs-DLLs (ADOX, SQLDMO, …) einen Einblick in die Objektstruktur einer Datenbank zu erhalten. Hiermit lassen sich unter anderem alle Tabellen, Views oder StoredProcedures einer Datenbank auflisten, aber auch Indizes, Keys und Relationen und noch vieles mehr. Gerade für Access ist diese Methode sehr angenehm, lassen sich auch wegen der kleinen Komplexität von Access nicht alle der Möglichkeiten einsetzen. Auf dem SQL-Server sind viele dieser Funktionen durch das Auslesen der Systemtabellen und –views nachzubilden, aber praktisch bleiben sie trotzdem, auch hier sind lange nicht alle Konstanten verwendbar und leider z.T. auch nicht von Microsoft dokumentiert, somit kann ich hierzu keinerlei Aussagen machen.

Auf alle möglichen Parameter möchte ich jetzt nicht eingehen da dies den Rahmen sprengen würde, aber die wichtigsten. Die Parameter welche bestimmen welche Objekte ausgelesen werden sollen werden durch das SchemaEnum definiert (welches ganze 41 Konstanten fast!).

Hier ein Auszug aus dem erwähnten Enum.

	BookmarkEnum

	Konstante
	Beschreibung

	adSchemaCatalogs
	Auslesen aller Datenbanken eines Servers. Da Access kein Server ist steht diese Funktion natürlich nicht zur Verfügung.

	adSchemaColumns
	Auslesen aller Spalten.

	adSchemaDBInfoLiterals
	Gesetzmäßigkeiten für Benamsungen der Datenbankobjekte.

	adSchemaForeignKeys
	Alle Fremdschlüssel der Datenbank auflisten.

	adSchemaIndexes
	Alle Indizes der Datenbank auflisten.

	adSchemaPrimaryKeys
	Alle Primärschlüssel einer Datenbank auflisten.

	adSchemaProcedureParameters
	Alle Parameter der StoredProcedures auslesen. Bei Access nicht verfügbar.

	adSchemaProcedures
	Alle StoredProcedures einer Datenbank auflisten. Bei Access werden auch Formular-Datengrundlagen ausgegeben. (Durch den Anfangsbuchstaben „~“ gekennzeichnet.)

	adSchemaProviderTypes
	Alle möglichen Felddatentypen und deren Beschaffenheit auslesen.

	adSchemaStatistics
	Derzeit aktuelle Anzahl der Zeilen aller Tabellenobjekte.

	adSchemaTableConstraints
	Primary- und ForeignKeys auflisten.

	adSchemaTables
	Alle Tabellen (und Views) einer Datenbank auflisten.

	adSchemaViews
	Alle Views einer Datenbank auflisten. Scheinbar nicht für SQL-Server verfügbar, kann jedoch leicht über die adSchemaTables-Auflistung gemacht werden.


Der Parameter Restrictions ist nie Pflicht, er kann gesetzt werden um einen Filter vorzudefinieren. Auf dem SQL-Server können hiermit auch Schemata anderer Datenbanken abgefragt werden. Hierbei wird ein Array mit den gewünschten Gegebenheiten und Beschränkungen übergeben bei dem leider kein 100%iges Schema erkennbar ist. Die Filterfunktion kann natürlich auch über das Recordset behandelt werden.

Der dritte Parameter ist nur nötig wenn providerspezifische Schemata abgefragt werden sollen. In diesem Fall muß die GUID des Schemas übergeben werden. Hierauf wird nicht näher eingegangen.

Hier ein paar Beispiele für den Gebrauch, die Daten werden hierbei direkt in ein DataGrid als Recordset (wird später beschrieben) übergeben.

Alle Tabellen (und Views/Auswahlabfragen) auflisten:

Set DataGrid1.DataSource = CnT.OpenSchema(adSchemaTables)

Tabellen des Datenbankbesitzers („dbo“) einer anderen Datenbank auflisten (SQL-Server):

Set DataGrid1.DataSource = CnT.OpenSchema(adSchemaTables, _

                        Array("DbName", "dbo"))

Alle Spalten einer Tabelle auflisten:

Set DataGrid1.DataSource = CnT.OpenSchema(adSchemaColumns, _

                        Array(Empty, Empty, "TabellenName"))

Alle Datenbanken eines Servers (SQL-Server) auflisten:

Set DataGrid1.DataSource = CnT.OpenSchema(adSchemaCatalogs)

Alle Indizes einer Tabelle auflisten:

Set DataGrid1.DataSource = Cn.OpenSchema(adSchemaIndexes, _

                        Array(Empty, Empty, Empty, _

                            Empty, "TabellenName"))

Beispiel für Connection

Um einige der genannten Eigenschaften und Methoden zu verdeutlichen, hier ein kleines Beispiel für das Verbinden auf einen SQL-Server und abschicken eines kleinen SQL-Statements:

Option Explicit

Dim Cn As ADODB.Connection

Private Sub Form_Load()

Dim i As Integer

Dim sMsg As String

On Error GoTo ErrHandle

Set Cn = New ADODB.Connection

With Cn

    ' Daten auf den Cursor ziehen

    .CursorLocation = adUseClient

    ' Datenbanktreiber

    .Provider = "SQLOLEDB.1"

    ' Servername

    .Properties("Data Source") = "FLO_MAIN"

    ' Datenbankname

    .Properties("Initial Catalog") = "db_Test"

    ' User-ID

    .Properties("User ID") = "florianr"

    ' Userpasswort

    .Properties("Password") = "flo"

'    ' NT-Authentifikation

'    .Properties("Integrated Security") = "SSPI"

    .Open

    .BeginTrans

    .Execute "DELETE FROM tblName WHERE FeldName = 'Hallo'"

    .CommitTrans

End With

Exit Sub

ErrHandle:

sMsg = "Es sind Fehler aufgetreten: " & vbCrLf & vbCrLf

With Cn

    .RollbackTrans

    For i = 0 To .Errors.Count - 1

        sMsg = sMsg & .Errors(i).Description & vbCrLf

    Next i

    .Errors.Clear

End With

MsgBox sMsg, vbCritical, "Datenbankfehler"

End Sub

Recordset

Das Recordset dient zur Anzeige und Navigation durch Datensätzen (Tabellen und Abfragen). Über Recordsets können Daten auch geändert, gelöscht oder angefügt werden. Für Frontend-Anwendungen sind Recordsets meist eines der wichtigsten Objekte. Die Eigenschaft CursorLocation ist wichtig, wird jedoch hier ausgelassen, da die Beschreibung bereits bei der Connection weiter oben zu finden ist.

ActiveConnection

Das Recordset kann nicht alleinstehend genutzt werden. Diese Eigenschaft bindet das Recordset an die Connection an. Es gibt ebenfalls die Eigenschaft ActiveCommand welche eine ähnliche Funktionalität über ein Command-Objekt bietet, jedoch mit beschnittener Funktionalität. Die Funktionalität des Recordsets beschränkt sich auf die maximal von der Connection erlaubten, es sind z.B. keine prinzipiell Änderungen möglich, wenn das die Connection nicht erlaubt.

Source

Source stellt ein SQL-Statement dar, welches angibt welche Daten abgefragt werden sollen. Beispiel:

Private Sub Form_Load()

RsT.ActiveConnection = Cn

RsT.Source = "SELECT * FROM tblName"

End Sub

AbsolutePosition

Diese Eigenschaft gibt die aktuelle Position in einem Recordset an. Der Wert ist ein Long-Wert. Dieser Wert kann sowohl abgefragt als auch gesetzt werden um durch das Recordset zu navigieren. Bei einem serverseitigem Cursor kann diese Eigenschaft nicht verwendet werden.

MoveFirst, MovePrevious, MoveNext, MoveLast, Move

Es ist zwar, wie beschrieben, auch mit der Eigenschaft AbsolutePosition möglich durch Datensätze zu navigieren, aber die eigentlichen Methoden hierfür sind die MoveX-Methoden.

MoveFirst, MovePrevious, MoveNext und MoveLast sollten sich selber erklären.

Move ist dafür gedacht, in Größeren Schritten zu navigieren. Hier kann man einen positiven oder negativen Long-Wert angeben um sich von einer aktuellen Position an um x Stellen nach vorne oder nach hinten zu bewegen. Der zweite Parameter der Move-Methode gibt an ab welcher Position gestartet werden soll.

	BookmarkEnum

	Konstante
	Beschreibung

	adBookmarkCurrent
	Default. Startet an der aktuellen Stelle.

	adBookmarkFirst
	Startet an der ersten Position im Recordset

	adBookmarkLast
	Startet an der letzten Position in Recordset


RecordCount

Der RecordCount gibt an wie viele Datensätze an den Client zurückgeliefert wurden. Diese Eigenschaft ist praktisch wenn man z.B. alle Datensätze mit einer Schleife durchlaufen möchten. Bei einem serverseitigem Cursor kann diese Eigenschaft nicht verwendet werden. 

Kleines Beispiel um einfach mal die aktuellen Positionen auszugeben:

Private Sub Form_Load()

Dim i As Long

Dim RsT As New ADODB.Recordset

With RsT

    .ActiveConnection = CnT

    .CursorLocation = adUseClient

    .Source = "SELECT * FROM tblName"

    .Open

    For i = 0 To .RecordCount - 1

        Debug.Print .AbsolutePosition

    Next i

End With

End Sub

CursorType

Der CursorType gibt die Funktionalität und Dynamik des Datensatz-Cursors. Der Cursor kann als Datensatzzeiger beschrieben werden. Er hat nichts mit dem Mauszeiger zu tun! Er gibt jedoch an wie flexibel durch die Datensätze navigiert werden kann.

	CursorTypeEnum

	adOpenForwardOnly
	Dieser Cursor ist der schnellste. Er verbraucht am wenigsten Speicher und Netzresourcen. Cursor kann jedoch nur vorwärts bewegt werden. Änderungen sind Zeilenweise möglich, jedoch nicht für andere User sichtbar.

	adOpenStatic 
	Ruft eine Kopie der abgefragten Daten ab. Daten können nicht geändert werden. Cursor in alle Richtungen beweglich.

	adOpenKeyset
	Dieser Cursor erstellt für jeden Datensatz einen Schlüssel. Er ist schneller als der Static- und der Dynamic-Cursor, da er nicht die ganzen Datensätze lädt sondern nur einen Schlüssel auf den Datensatz. Änderungen und löschen sind möglich und werden anderen Usern angezeigt. Einzig anfügen wird anderen Usern nicht angezeigt.

	adOpenDynamic
	Dieser Cursor ist der flexibelste, er ermöglicht alle Bewegungen. Alle Änderungen werden allen angezeigt. Er ist aber auch der ressourcenbeanspruchenste.


Der adOpenForwardOnly-Cursor ist zwar mit Abstand der schnellste, er ist aber nicht in der Lage zu einem hinter der aktuellen Position liegenden Datensatz zu springen. Er ist somit nur ratsam wenn nur Daten angefügt werden sollen, aber keine geändert oder angezeigt werden sollen.

Der statische Cursor bringt eine sehr große Funktionalität, ist aber der langsamste und lässt keine Änderungen zu, weil die Daten in einer Art Screenshot an den Client geliefert werden. Somit sind alle Daten auf dem Client, haben aber eigentlich keine Verbindung mehr mit der DB.

Der dynamische Cursor ist der funktionellste Cursor. Es werden alle Navigationsmöglichkeiten gestellt, Änderungen sind möglich und bei einem aktualisieren des Recordsets werden auch die Änderungen anderer User angezeigt. Dieser Cursor ist für SQL-Server nicht möglich, wenn er hier gesetzt wird löst ADO leider keinen Fehler aus sondern wählt selbst einen anderen Cursor (meist den statischen). Aus diesem Grund ist eher der nächste Cursor ratsam.

Der Keyset-Cursor ist der zweitschnellste Cursor und der mit der zweitgrößten Funktionalität. Er lässt jegliche Änderungen zu und zeigt auch die anderer User an. Einzig neu angefügte Datensätze von anderen Usern werden durch ein normales aktualisieren des Recordsets nicht angezeigt. Auf Grund der ersten genannten Beschaffenheiten bietet er sich am besten an im Gebrauch mit professionellen Datenbankanwendungen.

LockType

Gibt an wie die Datensatzsicherheit und die Änderungsberechtigungen des Recordsets zu setzten sind. Die höchste Zugriffsberechtigung des Recordsets orientiert sich jedoch an den eingestellten Berechtigungen der zugrundeliegenden Connection.

	LockTypeEnum

	Konstante
	Beschreibung

	adLockReadOnly
	Nur Lesen, Daten können nicht verändert werden (Default)

	adLockPessimistic
	Daten werden beim öffnen des Recordsets für andere User gesperrt.

	adLockOptimistic
	Daten werden erst beim Update gesperrt.

	adLockBatchOptimistic
	Dieser Modus wird nur für Stapelaktuallisierung benützt.


ReadOnly sollte sich selber beschreiben.

Pessimistischer LockType bedeutet, daß die Daten vom ersten User der sie aufruft für weitere Änderungen gesperrt werden. Die Daten können jedoch auch von anderen Usern angezeigt, jedoch nicht mehr geändert werden. Er ist ratsam wenn die Datenbankstruktur das Ändern von einzelnen Fällen vorsieht. Für Auflistungen ist er jedoch nicht empfehlenswert, da alle angezeigten Daten für nachfolgende User nicht mehr Änderbar sind.

Optimistisch bedeutet, daß die Daten von allen Usern mit vollem Zugriff benutzt werden können. Wenn ein aktuell geänderter Datensatz jedoch bereits von einem anderem User seit dem anzeigen geändert wurde kann ein Fehler auftreten.

Der adLockBatchOptimistic-Cursor ist für den Umgang mit Stapelverarbeitungen zu setzen. Hier werden bei Offline-Recordsets die geänderten Datensätze markiert und können in Nachhinein auf einmal alle mit der Datenbank abgeglichen werden. Er kann eine Datenbank sehr entlasten. Eine genauere Beschreibung des Einsatzes wird noch später beschrieben.

Fields

Diese Collection ist für Datenbearbeitung wohl die wichtigste des Recordsets. Über die Fields können erst die einzelnen Spalten und damit auch die Werte des Recordsets angesprochen werden.

Es gibt vier Möglichkeiten die Spalten des Recordsets anzusprechen:

MsgBox Rs.Fields("SpaltenName")

MsgBox Rs("SpaltenName")

MsgBox Rs.Fields(0)

MsgBox Rs!SpaltenName

Die letzte Schreibweise ist aber sehr unsauber und nicht ratsam! Da die Fields-Collection die Default-Eigenschaft des Recordsets ist kann man sich das Wort Fields sparen und die Schreibweise zwei nutzen. Bei der dritten Methode wird nicht der Name der Spalte angegeben sondern der Index in der Collection, sie bietet sich an um Beispielsweise über eine Schleife alle Spaltennamen zu erfahren.

Diese Collection bietet eine Unmenge an eigenen Methoden, Eigenschaften und auch Properties, wir möchten hier nur ein paar beschreiben:

Name

Gibt den Namen der Spalte an.

Count

Gibt die Anzahl der vorhandenen Spalten an. Hier ein kleines Beispiel um alle Spaltennamen des Recordsets in eine ListBox einzutragen:

Private Sub Form_Load()

Dim i As Long

With RsT

    .ActiveConnection = CnT

    .CursorLocation = adUseClient

    .Source = "SELECT * FROM TabellenName"

    .Open

    For i = 0 To .Fields.Count - 1

        List1.AddItem .Fields(i).Name

    Next i

End With

End Sub

Type

Diese Eigenschaft gibt an welches Spaltenformat für die aktuelle Spalte vorliegt. Diese Eigenschaft liefert einen Long-Wert zurück welcher eine Konstante des DataTypeEnums (Achtung, dieses Enum wird auch von DAO genutzt, daher Deklaration von Variablen immer über ADODB.DataTypeEnum!) zurückliefert.

	Ein paar der Konstanten dieses Enums (nicht alle)

	Konstante
	Beschreibung
	VB-Typ

	adBoolean
	Ja/Nein. Der Wert ist 0 für „nein“ und –1 für „ja“.
	Boolean

	adChar
	Text mit bis 255 Zeichen mit fester Länge (ASCII)
	String * 255

	adCurrency
	Wärung
	Currency

	adDate
	Datums-/Zeitwert.
	Date

	adDouble
	Fließkommazahl
	Double

	adInteger
	Große Nichtkommazahl
	Long

	adLongVarBinary
	Großes Binärfeld
	Alle Binärformate

	adLongVarChar
	Großer Text mit nichtfester Länge (ASCII)
	String

	adLongVarWChar
	Großer Text mit nichtfester Länge (UNICODE)
	String

	adSingle
	Kommazahl
	Single

	adSmallInt
	„Normale“ Nichtkommazahl
	Integer

	adTinyInt
	Kleine Nichtkommazahl
	Byte

	adVarChar
	Kleiner Text, bis 255 Zeichen mit nicht fester Länge (ASCII)
	String

	adVarWChar
	Kleiner Text, bis 255 Zeichen mit nicht fester Länge (UNICODE)
	String

	adWChar
	Text mit bis 255 Zeichen mit fester Länge (UNICODE)
	String * 255


DefinedSize

Eigenschaft für Textfelder. Gibt die maximale Länge des Textes an. ActualSize gibt die aktuelle Länge an.

OriginalValue

Gibt nach dem Ändern und vor dem Speichern des Feldes den Originalwert des Feldes an welcher bislang in dem Feld steht.

Value

Gibt den Wert des Feldes an. Diese Eigenschaft ist die Default-Eigenschaft der Fields und kann somit bei der Schreibweise weggelassen werden:

MsgBox Rs.Fields("SpaltenName")

' Entspricht

MsgBox Rs.Fields("SpaltenName")

BOF, EOF

Diese Eigenschaften geben einen Boolean-Wert zurück, welcher auf True steht, wenn entweder BOF (BeginnOfFile) oder EOF (EndOfFile) erreicht wurde. Die Werte sind erst dann True wenn über die erste/letzte Position hinaus navigiert wurde.

Bookmark

Bookmark ist ein Lesezeichen welches vom aktuellen Datensatz abgefragt werden kann oder zur Navigation gesetzt werden kann. Das Format ist Variant. Der Sinn liegt darin, daß man auch nach einem Filtern von Daten (weiter unten) zu der vorherigen Position springen möchte, da die AbsolutePosition nach einem Filter nicht mehr stimmt.

Cancel

Wie bei der Connection beschrieben hat diese Methode mit asynchronen Arbeiten zutun, Beispiele später.

CancelUpdate

Diese Eigenschaft setzt alle Originalwerte der Felder zurück bevor diese gespeichert wurden. CancelBatch entspricht dieser Funktion für Stapelverarbeitungen, hiermit können alle Änderungen wieder zurück gesetzt werden.

Clone

Die Clone-Metode ist sehr praktisch um einen zweiten, eigenständigen, Recordset mit den gleichen Attributen wie die des zugrundezulegenden zu erstellen.

Mit:

Set RsNeu = Rs

Würde kein neuer Recordset erstellt werden sondern nur ein Zeiger auf den Basis-Recordset. Somit würden sich alle Navigationen und anderen Aktionen auf Rs beziehen.

Daher:

Ser RsNeu = Rs.Clone

So kann RsNeu eigenständig als Recordset genutzt werden.

EditMode

Der EditMode gibt an in welchem Status sich ein Recordset derzeit befindet. Hierfür gibt es eigentlich nicht viel zu tun. Der Mode bleibt bestehen bis die Änderungen durch Update oder UpdateBatch vollendet werden. Es gibt 4 verschiedene Stati welche die durch das EditModeEnum definiert werden:

	EditModeEnum

	Konstante
	Beschreibung

	adEditNone
	Es liegt derzeit kein besonderer Mode vor.

	adEditInProzess
	Es werden gerade Änderungen an einem bestehenden Datensatz durchgeführt.

	adEditAdd
	Es wird gerade ein neuer Datensatz angelegt.

	adEditDelete
	Es wird gerade ein Datensatz gelöscht.


Filter

Diese Eigenschaft lässt kleine SQL-ähnliche Filter auf ein offenes Recordset zu ohne das Recordset zu schließen und neu öffnen zu müssen. Hier ein paar Beispiele:

Rs.Filter = "ReportId < 10 AND Titel LIKE 'E%'"

Rs.Filter = "Titel LIKE 'E%'"

Rs.Filter = "Vorname = 'Stefan' AND Nachname LIKE 'B%'"

Gelöscht wird der Filter mit:

Rs.Filter = adFilterNone

Find

Die Find-Methode ist zwar schneller als die Filter-Eigenschaft, lässt aber nur eindeutige Suchbegriffe mit „=“ (also nicht mit „LIKE“, „<“, „>“ oder „%“) zu. Es wird auch nur ein Datensatz, der erste passende, zurückgeliefert. Da es sich bei Find um eine Methode und nicht eine Eigenschaft handelt wird ohne „=“ gearbeitet:

Rs.Find = "ReportId = 10"

Zurückgesetzt wird die Find-Methode auch mit Rs.Filter = adFilterNone.

GetRows

Diese Methode kann genutzt werden um alle Werte des Recordsets in einen zweidimensionalen Varianten zu schreiben. Alle Änderungen können jedoch nicht mehr über eine ADO-Methode in das Recordset und somit nicht in die Datenbank zurückgegeben werden. Einsatz:

Private Sub Form_Load()

Dim VRs As Variant

With RsT

    .ActiveConnection = CnT

    .CursorLocation = adUseClient

    .Source = "SELECT * FROM Grunddaten"

    .Open

End With

VRs = RsT.GetRows

MsgBox VRs(0, 1)

End Sub

Bei dem Array gilt:

MsgBox VRs(SpaltenIndex, ZeilenIndex)

NextRecordset

Mit ADO ist es möglich in ein Recordset-Objekt mehrere Tabellen oder Abfragen gleichzeitig zu übergeben. Die einzelnen SQL-Selects müssen durch ein Semikolon „;“ getrennt werden. Bei Access ist dieses Verfahren nicht möglich. Hier ein Beispiel um die Daten aus zwei Tabellen gleichzeitig in ein Recordset zu holen und diese dann an ein anderes Recordset zu übergeben:

Private Sub Form_Load()

Dim RsT As New ADODB.Recordset

Dim Rs As ADODB.Recordset

cnSet

RsT.ActiveConnection = CnT

RsT.CursorLocation = adUseClient

RsT.Source = "SELECT * FROM tblTest1;SELECT * FROM tblTest2"

RsT.Open

Set Rs = RsT

MsgBox Rs.RecordCount

Set Rs = RsT.NextRecordset

MsgBox Rs.RecordCount

End Sub

AbsolutePage, PageCount, PageSize

Was haben Recordsets mit Pages (Seiten) zu tun? Man kann bei ADO, z.B. bei Ausdrucken, das Recordset in gewünschte Partitionen aufgliedern.

PageSize gibt an oder setzt die Größe der Seiten welche definiert werden sollen. Default ist 10.

PageCount gibt an wie viele Seiten auf Grundlage der PageSize und des RecordCounts vorhanden sind.

AbsolutePage gibt an oder setzt die aktuelle Seite im Recordset.

Requery

Fragt alle Daten des Recordsets komplett neu ab.

Rs.Requery

' Enspricht:

Rs.Close

Rs.Open

Wenn die Daten gerade geändert werden wird ein Fehler ausgelöst.

Resync

Diese Methode ist ähnlich der Requery-Methode. Auch hier werden die Daten mit denen der Datenbank neu abgeglichen. Sie ist vor allem im Gebrauch mit statischen Cursorn ist diese Methode recht hilfreich. Resync führt nicht Close-Open aus, damit ist es schneller als die Requery-Methode, dadurch werden aber keine neu angefügten Daten von anderen Usern angezeigt.

Save

Bei ADO ist es möglich mit sogenannten Offline-Recordsets zu arbeiten. Mit Save können Daten eines Recordsets auf dem Client physisch gespeichert. Der optionale Parameter PresistFormat gibt an in welchem Format die Daten gespeichert werden sollen. Wenn für das Recordset derzeit ein Filter besteht werden nur die aktuell gezeigten Daten gespeichert.

	PresistFormatEnum

	Konstante
	Beschreibung

	adPersistADTG
	Default. Speichert die Daten im ADTG-Format (Advanced Data Tablegram). Dies ist ein optimiertes Format

	adPersistXML
	Speichert die Daten im XML-Format


Sort

Diese Eigenschaft gibt an oder setzt ob und wie ein Recordset sortiert werden soll. Die Übergabe erlaubt mehrere Sortierungen gleichzeitig. Sortiert wer den kann entweder aufsteigend (ASC, Default) oder absteigend (DESC).

Wenn man z.B. im Recordset nach der Spalte Nachnamen sortieren will, jedoch bei gleichen Nachnamen den Datensatz des ältesten oben haben möchte, dann könnte die Sort z.B. so genutzt werden:

RsT.Sort = "Nachname ASC, Geburtsdatum DESC"

Update

Mit Update können die geänderten Daten in die Datenbank gespeichert werden. Wenn Daten geändert werden und dann navigiert wird (z.B. MoveNext) wird automatisch Update ausgeführt. UpdateBatch ist ähnlich aber für Stapelverarbeitungen gedacht.

AddNew

Über AddNew wird ein neuer Datensatz initialisiert. Übergeben werden die Daten mit Update:

Option Explicit

Dim RsT As New ADODB.Recordset

Private Sub Form_Load()

With RsT

    .ActiveConnection = CnT

    .CursorLocation = adUseClient

    .CursorType = adOpenKeyset

    .LockType = adLockOptimistic

    .Source = "SELECT * FROM tblTest"

    .Open

    .AddNew

    .Fields("Spalte1") = "Hallo"

    .Fields("Spalte2") = 123

    .Update

End With

End Sub

Offline-Recordsets

Offline-Recordsets sind eine neue Möglichkeit der Datenverarbeitung welche von ADO zur Verfügung gestellt wird.

Recordsets aus dem nichts

Es ist möglich ein Recordset komplett aus dem nichts zu erstellen und dieses dem Client z.B. als neue und extrem leistungsfähige Version von Collections und Arrays zu nutzen. (Dies darf nicht mit dem erstellen einer Tabelle auf einer Datenbank verwechselt werden, diese Recordsets sind rein auf dem Client vorhanden.)

Die Daten können nicht nur mit fast allen Methoden des Recordset-Objekts auf dem Client genutzt werden sondern auch physisch auf dem Client gespeichert und später wieder geöffnet werden.

Hier ein kleines Beispiel wie man eine Tabelle für eine kleine Personenverwaltung erstellt:

Option Explicit

Dim RsT As New ADODB.Recordset

Private Sub Form_Load()

With RsT

    .Fields.Append "Nachname", adVarChar, 255

    .Fields.Append "Vorname", adVarChar, 255

    .Fields.Append "Alter", adSmallInt

    .Open

    .AddNew

    .Fields("Nachname") = "Mustermann"

    .Fields("Vorname") = "Manfred"

    .Fields("Alter") = 25

    .Update

End With

End Sub

Diese Daten dann auf dem Client in eine Datei speichern:

RsT.Save "C:\Temp\Personen.dat"

Zu einem späterem Zeitpunkt können die Daten wieder über Open abgerufen werden (hierbei muß ActiveConnection = Nothing gesetzt sein und das Recordset natürlich geschlossen sein!):

Option Explicit

Dim RsT As New ADODB.Recordset

Private Sub Form_Load()

If RsT.State = adStateOpen Then RsT.Close

If Not RsT.ActiveConnection Is Nothing Then

    Set RsT.ActiveConnection = Nothing

End If

RsT.Open "E:\Temp\Personen.dat"

MsgBox RsT("Nachname")

End Sub

Stapelverarbeitung

Serverressourcen sind gerade bei sehr großen Anwendungen knapp. Es gibt bei ADO die sehr effiziente Möglichkeit der Stapelverarbeitung, das heißt die Connection wird hierfür vom Recordset getrennt und evtl. auch geschlossen. Es ist nun möglich nach belieben Änderungen an den Daten vorzunehmen und erst später auf einmal alle Daten mit der Datenbank abzugleichen. Hierfür werden die Batch-Attribute von ADO genutzt.

Beim verbinden wird hier der LockType auf adLockBatchOptimistic gesetzt um dem Recordset mitzuteilen, daß die Daten beim ändern durch ein Flag als „dirty“ markiert werden sollen, über dieses Flag weiß das Recordset später welche Daten geändert wurden und welche mit der Datenbank abzugleichen sind.

Öffnen eines Recordsets für Stapelverarbeitung und nachfolgendes schließen der gesamten Verbindungen zwischen Recordset, Connection und Server. Nach dieser Prozedur ist das Recordset komplett für den Client verfügbar die Verbindung zur Datenbank ist jedoch komplett geschlossen und somit ein User weniger der von der Datenbank aktuell verwaltet werden muß:

Option Explicit

Dim Rs As New ADODB.Recordset

Dim Cn As New ADODB.Connection

Private Sub Form_Load()

With Cn

    .Mode = adModeShareDenyNone

    .Provider = "Microsoft.Jet.OLEDB.4.0"

    .Properties("Data Source") = "F:\Test_DBs\Access\db3.mdb"

    .Open

End With

With Rs

    .ActiveConnection = Cn

    .CursorLocation = adUseClient

    .LockType = adLockBatchOptimistic

    .Source = "SELECT * FROM tblName"

    .Open

    ' Connection entfernen

    Set .ActiveConnection = Nothing

End With

' Connection zum Server schließen

Cn.Close

End Sub

Nachdem die Daten nun beliebig geändert, gelöscht und neu angelegt worden sind können alle Daten auf einmal mit der Datenbank abgeglichen werden:

' Die Verbindung zum Server wird neu hergestellt

' (Achtung bei UserId/-passwort muß je nach der Einstellung

' von "Presists Security Info" neu gesetzt werden)

Cn.Open

' Recordset wird wieder verbunden

Set Rs.ActiveConnection = Cn

' Die Änderungen werden auf einmal an die DB übergeben

Rs.UpdateBatch

Um alle Änderungen rückgängig zu machen kann man einfach ein Stapel-Cancel ausführen:

Rs.CancelBatch

Command

Das Command-Objekt ist das dritte wichtige ADO-Objekt. Mit diesem kann man alle Arten von „Aktions-SQL-Statements“ ausführen (z. B. UPDATE, CREATE TABLE, INSTERT INTO, ...). Einzelne SQL-Statements können jedoch auch über die Connection mit der Methode Execute abgeschickt werden. Die eigentlich Kraft dieses Objektes kommt erst dann zum Einsatz wenn man mit StoredProcedures arbeitet, was auch mit Access in einem kleinem Maß möglich ist.

Im Gegensatz zu dem Recordset-Objekt kann man das Command optional entweder an eine Connection anbinden oder die Verbindung selbst erstellen.

Die Programmierer die den Umgang mit professionellen Datenbankservern bereits gewöhnt sind werden den Umgang mit StoredProcedures bereits kennen. Bei Access stellt sich vielleicht die Frage „Was und wie soll ich hier mit StoredProcedures und was ist das überhaupt?“. 

Was sind StoredProcedures? StoredProcedures sind auf der Datenbank platzierte, oft sehr große, SQL-Routinen. Der Sinn liegt in vier Punkten. Erstens sind DB-Server meist wesentlich Leistungsfähiger als die zugreifenden Clients, somit können die Aktionen wesentlich effizienter ausgeführt werden. Zweitens durch sinnvoll aufgebaute StoredProcedures wird sehr viel Traffic zwischen der Datenbank und dem Client (der Datenbankanwendung) vermieden, das ist für das Netzwerk gut und bringt starke Performanceverbesserungen. Als Drittes kann durch StoredProcedures der meiste SQL-Code aus dem FrontEnd gebracht werden sodass sich die Anwendungsentwicklung wirklich nur mit der Anwendung beschäftigen muß. Der letzte nicht unerhebliche Grund ist, wenn der, für einen Datenbankentwickler grauenhafte (!), Fall eintritt, daß die Datenbankstruktur geändert werden muß ist es bei StoredProcedures nicht unbedingt nötig auch gleich die gesamte Anwendung neu zu schreiben sondern nur die Lokalen Prozeduren umzuschreiben.

Was mache ich in Access damit? In Access sind leider nur sehr bedingt größere SQL-Scripts möglich, da leider keine Variablen deklariert werden können, keine Cursor und Schleifen verfügbar sind und nicht mehrere Aktionen auf einmal ausgeführt werden können. Prinzipiell sind aber alle Nichtauswahlabfragen (also Löschabfragen, Anfügeabfragen, Pivot-Tabellen, …) als StoredProcedures zu betrachten. Durch ein paar Übergabeparameter kann man sich hier Slim-Prozeduren schreiben. In den folgenden Beispielen werden übrigens Übergabeparameter immer mit einem vorgestellten @-Zeichen geschrieben, das ist bei Access-Parametern zwar nicht nötig, aber beim SQL-Server so vorgeschrieben und wir wollen auch bei Access diese Schreibweise beibehalten.

Verbindung mit einer Datenbank

Es gibt, wie schon beschrieben, zwei Möglichkeiten ein Command-Objekt mit einer Datenbank zu verbinden.

Erstellen mit bestehender Connection

Dies ist die meistens sinnvolle Methode um ein Command mit einer DB zu verbinden. Es wird einfach mit einer bestehenden Connection verbunden:

Dim Cmd As New ADODB.Command

Cmd.ActiveConnection = Cn

Erstellen ohne Connection

Statt einer Connection kann auch ein String übergeben werden welcher dem ConnectionString gleich zusetzten ist:

Dim Cmd As New ADODB.Command

Cmd.ActiveConnection = "Provider=Microsoft.Jet.OLEDB.4.0;" & _

  "Data Source=F:\Test_DBs\Access\db3.mdb;" & _

  "Jet OLEDB:Database Password=test;" & _

  "Jet OLEDB:System database=F:\Test_DBs\Access\db3.mdw;" & _

  "User ID=florianr;" & _

  "Password=flo"

Execute

Die Execute-Methode ist ähnlich der des Connection-Objektes in seiner Grundfunktionalität. Es führt ein SQL-Statement aus. Bei dem Command-Objekt ist Execute jedoch wesentlich stärker. Der SQL-String wird nicht angehängt sondern in dem meisten Fällen über viele Parameter und Eigenschaften zusammengestellt. Ein wichtiger Parameter ist RecordsAffected, dieser ist kein Übergabeparameter sondern ein Rückgabeparameter. Bei Aktionen auf der Datenbank (löschen, ändern, hinzufügen) wird hier die Anzahl der betroffenen Datensätzen in eine übergebene Long-Variable eingetragen. Als Objekt liefert Execute, wie die Connection, einen Recordset zurück.

CommandType

Der CommandType gibt an was ausgeführt werden soll. Dafür liegt das CommandTypeEnum zu Grunde.

	CommandTypeEnum

	Konstante
	Beschreibung

	adCmdText
	Es wird ein Text übergeben welcher z.B. ein SQL-Statement oder eine StoredProcedure (mit Parametern!) übergibt.

	adCmdTable
	Es wird als Text ein Tabellenname übergeben, somit wird ein Recordset zurückgeliefert welches die Tabelle 1:1 wiederspiegelt. (Intern wird ein SQL-Statement erstellt)

	adCmdTableDirect
	Es wird als Text ein Tabellenname übergeben, somit wird ein Recordset zurückgeliefert welches die Tabelle 1:1 wiederspiegelt. (Es wird kein SQL-Statement erstellt)

	adCmdStoredProc
	Es wird der Name einer StoredProcedure übergeben.

	adCmdUnknown
	Default. Es wird nicht festgesetzt was als CommandText gesetzt wird.

	adCmdFile
	Liefert ein Recordset zurück welches lokal auf dem Client als Datei gespeichert ist.

	adExecuteNoRecords
	Es werden explizit keine Daten an den Client zurückgeliefert. Vor allem beim Gebrauch mit StoredProcedures kann hier eine Datenrückgabe unterdrückt werden.

Kombination mit adCmdText oder adCmdStoredProc.


CommandText

Der CommandText gibt einen String an welcher z.B. einen Tabellennamen, ein SQL-Statement oder den Namen einer StoredProcedure darstellt.

Ein paar Beispiele:

Öffnen einer Tabelle

Dim Cmd As New ADODB.Command

Cmd.ActiveConnection = Cn

Cmd.CommandType = adCmdTable

Cmd.CommandText = "tblName"

Set Rs = Cmd.Execute()

Ausführen einer StoredProcedure (ohne Parameter mit Recordset als Rücklieferung):

Dim Cmd As New ADODB.Command

Dim Rs As ADODB.Recordset

Cmd.ActiveConnection = Cn

Cmd.CommandType = adCmdStoredProc

Cmd.CommandText = "sp_ProcName"

Cmd.Execute

Ausführen eines normalen SQL-Statements (Anzahl der gelöschten Datensätze wird zurückgeliefert):

Dim Cmd As New ADODB.Command

Dim i As Long

Cmd.ActiveConnection = Cn

Cmd.CommandType = adCmdText

Cmd.CommandText = "DELETE FROM tblName WHERE Alter > 80"

Cmd.Execute i

MsgBox i

CreateParameter

Diese Methode ist vor allem für Access-Entwickler eine geniale Erfindung! Wenn man die Parameter einer StoredProcedure nicht von der DB abfragen will, oder bei Access leider nicht kann, bietet diese Funktion eine sensationelle Selbsthilfe. Mit dieser Funktion ist es möglich für das Command selbst Parameter lokal zu erstellen und diese anzuhängen.

Die Syntax erwartet bis zu fünf Übergabeparameter:

Name

Name des zu erstellenden Parameters

Type

Art des zu übergebenden Parameters. Hierfür liegt das DataTypeEnum zu Grunde, eine genauere Beschreibung ist im Kapitel Recordset unter Fields-Type zu finden.

Direction

Gibt an in welche Richtung der Parameter agiert. Die Konstanten stammen aus dem ParameterDirectoinEnum. Bei Access sind mir leider nur Inputparameter bekannt.

	ParameterDirectoinEnum

	Konstante
	Beschreibung

	adParamUnknown
	Die Richtung des Parameters ist nicht bekannt.

	adParamInput
	Es handelt sich um einen Input-Parameter

	adParamOutput
	Es handelt sich um einen Output-Parameter

	adParamInputOutput
	Der Parameter ist sowohl Übergabeparameter, liefert aber auch einen Wert zurück.

	adParamReturnValue
	Beschreibt einen Return-Wert


Size

Diese Eigenschaft ist wie die gleichnamige Eigenschaft des Field-Objektes des Recordsets zu betrachten. Es ist eigentlich nur bei Text-Parametern nötig und gibt die maximale Länge des zu übergebenden Wertes an.

Value

Hier wird der eigentlich zu übergebende Wert eingetragen.

Ein Beispiel für den Einsatz dieser Funktion wird in der Beschreibung der Parameter geliefert.

Parameters

Diese Collection macht das Command erst zu dem was es ist, ein mächtiges Instrument im Umgang mit professionellen Datenbankanwendungen. Hier werden die Parameter für StoredProcedures (Aktionsabfragen in Access) angehängt.

Der sinn von StoredProcedures ist es ja möglichst dynamisch auf gewünschte Aktionen des Users einzugehen. Dafür werden Übergabeparameter benötigt und diese werden hier angehängt.

Refresh

Reine Access-Programmierer müssen die Beschreibung dieser sehr angenehmen Methode leider überspringen, Access bietet diese Funktionalität leider nicht.

Die Refesh-Methode verbindet sich kurz mit dem Server und fragt für die vorher angegebene Prozedur die Parameter ab und erstellt die Collection der Parameters neu. Dies verursacht zwar einen minimalen kurzen Verkehr im Netzwerk, welcher aber in keiner Relation steht zum lokalen erstellen der Parameter.

Append

Wer auf Refresh verzichten will, oder muß kann auch lokal die zu übergebenden Parameter erstellen und an das Command anhängen. Hierbei wird ein ADO-Parameter-Objekt angehängt.

Beispiele für Parameter

Ein Command mit Refresh und Rücklieferung eines OUTPUT-Parameters ausführen:

Private Sub Form_Load()

Dim Cmd As New ADODB.Command

With Cmd

    .ActiveConnection = CnT

    .CommandType = adCmdStoredProc

    .CommandText = "sp_New_Adress"

    .Parameters.Refresh

    .Parameters("@Nachname") = "Mustermann"

    .Parameters("@Vorname") = "Manfred"

    .Parameters("@Telefon") = "012-456789"

    .Execute

    MsgBox .Parameters("@AdressId")

End With

End Sub

Ein Command mit CreateParameter und Recordset als Rückgabe (für Access sehr wichtig!):

Private Sub Form_Load()

Dim Cmd As New ADODB.Command

Dim RsBestell As ADODB.Recordset

With Cmd

    .ActiveConnection = CnT

    .CommandType = adCmdStoredProc

    .CommandText = "sp_Search_Bestellungen"

    .Parameters.Append .CreateParameter("@BestellDatum", _

                            adDate, adParamInput, , Date)

    Set RsBestell = .Execute

End With

End Sub

Bei dem letzten Beispiel könnte bei Access beispielsweise ein großes SELECT-Statement hinterlegt sein welches über verschiedene JOINs eine Tabelle Bestell_Main mit einer Tabelle Bestell_Detail, einer Tabelle Artikel und einer Tabelle Einkaeufer verbindet.

Stream

Dieses Objekt ist eine Neuerung seit ADO 2.5. Es ist vor allem für den Gebrauch mit Internet-Datenbanken und XML bzw. ASP wichtig, auf was ich leider nicht genauer eingehen kann. Außerdem kann er sehr ähnlich dem Stream-Objekt aus der Microsoft Scripting Runtime verwendet werden, was ich hier auch weg lassen möchte.

Einzig hier zu erwähnen ist der Einsatz mit Binären (Bilder, Dokumente, …) Daten in einem Recordset. Diese Daten können nämlich nicht einfach so zugewiesen werden:

Me.Picture1.Picture = Rs.Fields("Bild”).Value

Im Geschäftsgebrauch hat es sich zwar im allgemeinen eingebürgert zur Entlastung der Datenbank keine Binärdaten in einer Datenbank zu speichern sondern nur Links auf die Dateien in einem Filesystem, aber Ausnahmen bestätigen ja bekanntlich die Regel. Gerade bei Access ist von speichern von Binären Daten klar abzusehen da hier sonst schnell die Grenze der zu verwaltenden Daten erreicht ist. Ein möglicher Einsatz währen Dokumentum-Systeme welche Faxe, Urkunden oder andere empfindlichen Daten verwalten.

Als Spaltenformat muß in der Datenbank logischerweise auch ein binäres Format vorliegen. Bei Access ist das das „OLE-Objekt“, auf dem SQL-Server ist es beispielsweise ein Image-Feld.

Datei in Recordset

Die Handhabung des Streams ist eigentlich recht einfach. Zum einlesen einer Datei wird das Objekt instanziiert und dann über LoadFromFile gefüllt. Wichtig ist, daß das Objekt nicht einfach mit Feld = Objekt überbeben werden kann sondern bei der Übergabe an das Binärfeld des Recordsets mit der Funktion Read ausgelesen werden muß.

Hier ein kleines Beispiel zum Speichern eines Bildes in einer DB:

Option Explicit

Dim Cn As ADODB.Connection

Dim Rs As ADODB.Recordset

Dim adStr As ADODB.Stream

Private Sub Command1_Click()

Dim sFile As String

sFile = "E:\Temp\test.jpg"

    Rs.AddNew

    If Dir$(sFile) = vbNullString Then

        MsgBox "Objekt kann nicht gefunden werden."

        Exit Sub

    End If

    ' ADO-Stream aus Datei einlesen

    adStr.LoadFromFile "E:\temp\test.jpg"

    ' ADO-Stream an DB übergeben

    Rs.Fields("FeldO").Value = adStr.Read

    Rs.Update

End Sub

Private Sub Form_Load()

Set Cn = New ADODB.Connection

With Cn

    .CursorLocation = adUseClient

    .Mode = adModeShareDenyNone

    .Provider = "Microsoft.Jet.OLEDB.4.0"

    .Properties("Data Source") = "F:\Test_DBs\Access\db3.mdb"

    .Open

End With

Set Rs = New ADODB.Recordset

With Rs

    .ActiveConnection = Cn

    .CursorLocation = adUseClient

    .LockType = adLockOptimistic

    .CursorType = adOpenKeyset

    .Source = "SELECT * FROM tblMyOle"

    .Open

End With

' ADO-Stream definieren

Set adStr = New ADODB.Stream

With adStr

    ' Festlegen, daß ein Binäres Format gelesen wird

    .Type = adTypeBinary

    .Open

End With

End Sub

Recordset in Datei

Eigentlich ist die andere Richtung einfach das selbe nur umgekehrt. Hier wird erst das Stream-Objekt mit Read aus dem Recordset ausgelesen und dann mit SaveToFile in eine Datei gespeichert:

Private Sub Command2_Click()

Dim sFile As String

Dim sMsg As String

sFile = "E:\Temp\testNeu.jpg"

If Dir$(sFile) <> vbNullString Then

    sMsg = "Die Datei ist bereits vorhanden. " & _

            "Soll die Datei gelöscht werden?"

    If MsgBox(sMsg, vbExclamation + vbOKCancel) = vbOK Then

        Kill sFile

    Else

        MsgBox "Datei konnte nicht ausgelesen werden."

        Exit Sub

    End If

End If

    Rs.MoveLast

    ' Stream aus DB-Feld einlesen

    adStr.Write Rs.Fields("FeldO")

    ' Stream als Datei speichern

    adStr.SaveToFile "e:\temp\test.jpg"

End Sub

Sonstige ADO-Methoden

Es sind nun eigentlich alle Objekte von ADO beschrieben. Es bleiben jedoch noch zwei komplett neue Methoden offen welche in den vorherigen Beschreibungen ausgelassen wurden. Gründe hierfür sind die teilweise Komplexität der Themen und die Tatsache das hierfür alle ADO-Objekte bekannt sein sollten.

UDL-Dateien

UDL-Dateien („Universal Data Links“) sind eine Andere Möglichkeit eine Connection zu erstellen und können den ConnectionString ersetzen. Ihr großer Vorteil liegt darin, daß bei Änderungen in der Serverlanschaft, der DB oder sonstiger grundlegender Eigenschaften der Datenbasis nicht ein komplett neues Kompilat (EXE) der Anwendung erstellt und herausgegeben werden muß sondern nur die extern liegende Datei ersetzt oder geändert werden muß was leicht durch eine kleine Batch-Datei gemacht werden kann.

Erstellen Sie hierfür im Explorer einfach eine neue Datei mit der Endung *.udl und öfnen diese mit einem Doppelklick. Dann können alle Eigenschaften für die Verbindung eingestellt werden und auch getestet werden. Geben Sie jedoch auf keinen Fall (!!) das Passwort mit ein und lassen dieses speichern, eine UDL ist zwar keine Textdatei kann jedoch ohne weiteres mit jedem Editor geöffnet werden! Das ist auch der eigentlich eigentliche Nachteil der Dateien, da der User diese Dateien leicht ändern kann, aber sein wir mal ehrlich, welcher User schaut sich alle Dateien an die bei einem Programm im Programmordner sind? Und außerdem, na wenn schon, er kann mit den Daten eigentlich nicht viel damit anfangen. (Bei Access sicherstellen, daß durch Änderungen des Zugriffsmodus keine zu großen Zugriffsberechtigungen entstehen können, hierfür die Userberechtigungen der System-MDW anpassen).

Wenn Sie sich eine UDL-Datei selbst mal mit einem Editor anschauen werden Sie sehen, daß die Daten darin eigentlich genau dem ConnectionString entsprechen. So können Änderungen auch sehr leicht über eine Batch-Datei durchgeführt werden.

Aus einem mir nicht verständlichen Grund scheint der Einsatz von UDL-Dateien leider nicht mit der Prompt-Eigenschaft der Connection vereinbar zu sein, also entweder nutzt man den Luxus des Microsoft-Anmeldefensters oder den einer UDL-Datei. Außer man verlässt sich auf die NT-Sicherheit. Einzige Möglichkeit dieses Manko zu umgehen ist eine kleine Fehlerroutine.

Hier ein kleines Beispiel wie diese Datei an eine Connection übergeben wird wenn NT-Authentifikation eingestellt ist:

Private Sub Form_Load()

CnT.Open "FILE NAME=" & App.Path & "\DBLnk.udl"

End Sub

Ein Beispiel wie man das Prompt trotz UDL nutzen kann:

Private Sub Command1_Click()

Set Cn = New ADODB.Connection

On Error GoTo ErrCn

Cn.Open "FILE NAME=E:\Temp\aaa.udl"

Exit Sub

ErrCn:

' Logindaten nachtragen

If Err.Number = -2147217843 Then  ' Loginfehler

    Cn.Properties("Prompt") = adPromptCompleteRequired

    Cn.Open

    Resume Next

Else

    MsgBox Err.Description

End If

End Sub

DataLinks

Dieses Objekt halte ich persönlich für Unsinn, aber der Vollständigkeit wegen soll es erwähnt werden. Es kommt aus einer anderen Bibliothek und einen neuen Verweis im Projekt auf die MSDAC („Microsoft OLE DB Service Component 1.0 Type Library“) setzen. Diese Library stellt eigentlich nichts anderes als eine Möglichkeit für das Erstellen einer Verbindung über den UDL-Dialog im Programm zur Verfügung. Auf Grund der Menge der Möglichkeiten welche eingestellt werden können sehe ich als einzige Möglichkeit der Verwendung einen SQL-Editor oder ähnliches. Wenn sie jedoch mehr Sinn in diesen Objekte sehen, bitte.

Wir sehen uns hier nur ein Objekt dieser Bibliothek an, das DataLinks-Objekt. Es gibt zwei Möglichkeiten den Dialog zu öffnen, entweder mit PromptEdit oder PromptNew, wobei sich diese im sinnvollen Einsatz wiederum eigentlich nur durch die Schreibweise unterscheiden. Bei PromptEdit wird eine bestehende Connection mit eventuellen vorgegebenen Einstellungen (z.B. über GetSettings) übergeben wird.

Beispiel für PromptEdit:

Private Sub Form_Load()

Dim dlk As New MSDASC.DataLinks

CnT.Provider = "SQLOLEDB.1"

CnT.Properties("Data Source") = "FLO_MAIN"

CnT.Properties("Initial Catalog") = "db_Test"

dlk.PromptEdit CnT

CnT.Open

End Sub

Beispiel für PromptNew:

Private Sub Form_Load()

Dim dlk As New MSDASC.DataLinks

Set CnT = dlk.PromptNew

CnT.Open

End Sub

Events

Seit ADO haben die Connection und das Recordset zusätzlich zu ihren Eigenschaften, Methoden und Funktionen auch Events. Diese Neuheit bietet die Möglichkeit Sicherheitsmechanismen, Fehlerbehandlungen und andere Akitonen zentral zu verwalten, sowie viele andere Möglichkeiten die bei DAO einfach nicht möglich waren. Auch im Bezug auf Arbeiten mit verknüpften Daten und vielem mehr sind diese Events sehr stark einzusetzen und sparen viel Code.

Zu beachten ist hierbei, daß die Objekte mit WithEvents im allgemeinen Bereich deklariert werden müssen und nicht von vorn herein mit New klassifiziert werden können sondern erst dynamisch (z.B. im Form_Load) mit Set neu instanziiert werden können:

Dim WithEvents Rs As ADODB.Recordset

' ...

Set Rs = New ADODB.Recordset

' ...

Die große Stärke der Events wird jedoch erst im Umgang mit asynchronen Abläufen ersichtlich. Hierzu folgen noch Beschreibungen weiter unten im Text.

Connection

Bei der Connection können Events beispielsweise genutzt werden um Fehlerroutinen global zu behandeln oder Stati von der Verbindung zur Datenbasis anzugeben. Man kann hier allgemeine Meldungen an den User zurückliefern oder auch aus Sicherheitsgründen Datenänderungen wieder zurücksetzen.

Ein kleines Beispiel für eine zentrale Behandlung um eventuelle Fehler in der DB zu verhindern:

Private Sub Cn_ExecuteComplete( _

        ByVal RecordsAffected As Long, _

        ByVal pError As ADODB.Error, _

        adStatus As ADODB.EventStatusEnum, _

        ByVal pCommand As ADODB.Command, _

        ByVal pRecordset As ADODB.Recordset, _

        ByVal pConnection As ADODB.Connection)

    If adStatus = adStatusErrorsOccurred Then

        MsgBox "Es ist ein Fehler aufgetreten."

        Cn.RollbackTrans

    Else

        Cn.CommitTrans

    End If

End Sub

Private Sub Cn_WillExecute( _

        Source As String, _

        CursorType As ADODB.CursorTypeEnum, _

        LockType As ADODB.LockTypeEnum, _

        Options As Long, _

        adStatus As ADODB.EventStatusEnum, _

        ByVal pCommand As ADODB.Command, _

        ByVal pRecordset As ADODB.Recordset, _

        ByVal pConnection As ADODB.Connection)

    Cn.BeginTrans

End Sub

Private Sub Form_Load()

Dim i As Integer

Dim Cmd As ADODB.Command

Set Cn = New ADODB.Connection

With Cn

    .CursorLocation = adUseClient

    .Provider = "SQLOLEDB.1"

    .Properties("Data Source") = "FLO_MAIN"

    .Properties("Integrated Security") = "SSPI"

    .Properties("Initial Catalog") = "db_Test"

    .Open

End With

Set Cmd = New ADODB.Command

    With Cmd

        .ActiveConnection = Cn

        .CommandType = adCmdStoredProc

        .CommandText = "sp_Test"

        .Parameters.Refresh

        .Parameters("@ReportId") = 6

        .Execute

    End With

End Sub

Beim ausführen einer StoredProcedure können auf dem Server durchaus erst Fehler auftreten wenn die ersten Änderungen bereits durchgeführt wurden. Über das Abfragen und eventuelle zurücksetzen der Änderungen kann man hier Fehleinträge auf der Datenbank vermeiden.

Recordset

Bei Recordsets können durch die verschiedenen Events gestimmte Aktionen in der Benutzeroberfläche ausgeführt werden und so Fehler vermieden werden. So kann beispielsweise beim klicken einer Schaltfläche „Neuer Datensatz“ und das Ausführen eines Rs.AddNew im MoveComplete die Schaltfläche aufgrund des EditModes disabled werden bis der Status nicht mehr adEditAdd ist. Dies verhindert Fehler durch nochmaliges klicken der Schaltfläche.

Oft bestehen die Daten einer Benutzeroberfläche nicht nur aus einem Recordset sondern zusätzlichen Detaildaten die beispielsweise in einem Grid angezeigt werden. Ein häufiger Einsatz ist hier ein Bestellformular wobei die Hauptdaten aus dem Lieferanten, Bestelldatum, Sachbearbeiter, usw. bestehen und die Detaildaten aus den verschiedenen Artikeln, Bestellmenge, Preisen, … .

Hier ein Beispiel für den Einsatz mit Haupt- und Detaildaten. Die Hauptdaten sind durch den Recordset RsMain dargestellt, die Detaildaten durch RsDetail. Die Verbindung der Daten durch das in beiden Recordsets vorhandene Feld „A_Id“ (Artikelnummer) gesichert. Jetzt wird beim Neuanlegen von Datensätzen die ID an die Detaildaten automatisch übergeben:

Option Explicit

Dim Cn As ADODB.Connection

Dim RsMain As ADODB.Recordset

Dim WithEvents RsDetail As ADODB.Recordset

Dim bChanged As Boolean

Private Sub Form_Load()

Set Cn = New ADODB.Connection

With Cn

    .CursorLocation = adUseClient

    .Provider = "SQLOLEDB.1"

    .Properties("Data Source") = "FLO_MAIN"

    .Properties("Integrated Security") = "SSPI"

    .Properties("Initial Catalog") = "db_Test"

    .Open

    Set RsMain = .Execute("SELECT * FROM tbl_Artikel")

End With

Set RsDetail = New ADODB.Recordset

With RsDetail

    .ActiveConnection = Cn

    .CursorLocation = adUseClient

    .CursorType = adOpenKeyset

    .LockType = adLockOptimistic

    .Source = "SELECT * FROM tbl_Bestellungen"

    .Open

End With

Set Me.DataGrid1.DataSource = RsDetail

End Sub

Private Sub RsDetail_FieldChangeComplete( _

        ByVal cFields As Long, _

        ByVal Fields As Variant, _

        ByVal pError As ADODB.Error, _

        adStatus As ADODB.EventStatusEnum, _

        ByVal pRecordset As ADODB.Recordset)

    bChanged = False

End Sub

Private Sub RsDetail_WillChangeField( _

        ByVal cFields As Long, _

        ByVal Fields As Variant, _

        adStatus As ADODB.EventStatusEnum, _

        ByVal pRecordset As ADODB.Recordset)

With RsDetail

    If .EditMode = adEditAdd Then

        If Not bChanged Then

            bChanged = True

            .Fields("A_Id") = RsMain.Fields("A_Id")

        End If

    End If

End With

End Sub

Hierarchische Recordsets

Hierarchische Recordsets sind eine Neuerung von ADO die es vorher einfach nicht gab. Hierbei handelt es sich um Recordsets welche in einer Baumstruktur an den Client zurückgeliefert werden. Früher konnte man von einander Abhängige Recordsets nur über zwei oder mehr Recordsets realisieren.

Man hat beispielsweise ein Recordset RsArtikelart und ein Recordset RsArtikelDetail. Beide werden in jeweils einem DataGrid (für DAO-ler: Nachfolger des DBGrid) angezeigt. Nun kann beispielsweise im Event RowColChange des DataGrids oder im Event MoveComplette des RsArtikelart-Recordsets die Daten für das Recordset RsArtikelDetail neu mit der Id der Artikelart abgefragt werden, so erhält man immer die Detaildaten zu einem Hauptdatensatz.

Eine andere Möglichkeit wäre über ein normales JOIN die Daten mit einander zu verknüpfen, jedoch ist hier eine schöne Navigation durch die Hauptwerte nur sehr unschön möglich.

Die hier verwendeten SQL-Statements werden nicht in VB-Code eingebunden gezeigt. Da es bei Statements für hierarchische Recordsets meist um größere Strings handelt überlasse ich diese Arbeit gerne jedem selbst. Ein kleiner Tipp, für Tests ist ein Formular mit einer TextBox für den SQL-String, einem „Microsoft Hirarchical FlexGrid“ und einem Button um das Recordset zu öffnen sehr hilfreich. 

MSDataShape

Der Aufbau eines Hierarchischen Recordsets erfordert eine gewisse Vorarbeit. Die normalen DB-Treiber von Access und SQL-Server (ODBC und auch OLEDB) schaffen diese Funktionalität nicht. Dafür gibt es von Microsoft den ADO-Treiber „MSDataShape“, welcher diese Funktionalität unterstützt, er sollte im Lieferumfang der MDAC_TYP vorhanden sein. Als erstes muß die Connection geändert werden. Der neue Treiber muß gesetzt werden und der eigentliche Datenbanktreiber nimmt die untergeordnete Rolle des „Data Provider“ ein.

Für Access:

With CnT

    .CursorLocation = adUseClient

    .Provider = "MSDataShape"

    .Properties("Data Provider") = "Microsoft.Jet.OLEDB.4.0"

    .Properties("Data Source") = "E:\Temp\db3.mdb"

    .Open

End With

Für SQL-Server:

With CnT

    .CursorLocation = adUseClient

    .Provider = "MSDataShape"

    .Properties("Data Provider") = "SQLOLEDB.1"

    .Properties("Data Source") = "FLO_MAIN"

    .Properties("Initial Catalog") = "db_Test"

    .Properties("User Id") = "florianr"

    .Properties("Password") = "flo"

    .Open

End With

Providerspezifische Eigenschaften wie z.B. „Jet OLEDB:Database Password“ werden in Form eines kleinen ConnectionStrings in die Eigenschaft „Extendet Properties“ übergeben.

Microsoft Hierarchical FlexGrid

Wer bereits mit ADO und den dazugehörigen Steuerelementen zu tun hatte ist sicher schon auf das „Microsoft Hierarchical FlexGrid“ (MSHFlexGrid) gestoßen. Abgesehen vom unaussprechbarem Namen ist das MSHFlexGrid unter anderem der Nachfolger des altbekannten FlexGrids. Das MSHFlexGrid ist aber auch das einzige Control das hierarchische Datensätze anzeigen kann. Wenn hier ein hierarchisches Recordset über die gewohnte Methode mit

Set MSHFlexGrid.DataSource = Rs

angebunden wir erscheinen am Anfang der Zeilen kleine „+“ oder „-„ Zeichen welche die Detaildaten ein- oder ausblenden können.

Grundstruktur

Der Aufbau eines hierarchischen Recordsets geschieht über ein SQL-Statement welches einige neue Begriffe und eine neue Syntax mit sich bringt. Den Anfang bildet ein SHAPE welches die obere Hierarchie des Recordsets angibt und einen Namen erhält (hier ArtikelHierarch). Dann wird über die APPEND-Methode eine Unterebene für die erste Hierarchieebene gebildet welches mit RELATE über den Namen der zu verbindenden Spalten angebunden wird.

SHAPE {SELECT * FROM tbl_Artikel ORDER BY A_Name} AS ArtikelHierarch

APPEND ({SELECT * FROM tbl_Bestellungen} AS BestellSub

RELATE 'A_ID' TO 'A_ID') AS BestellSub

Wenn man nun eine Tabelle tbl_Artikel hat mit den Daten über alle Artikel und eine Tabelle tbl_Bestellungen mit allen Bestellungen die für die Artikel und die Artikel-ID A_ID kann man nun alle Artikel und die jeweils dazu bestehenden Bestellungen anzeigen:

Zu beachten ist bei dieser Syntax die vielen neuen Schreibweisen. Die einzelnen Recordsets werden hier mit geschweiften Klammern „{…}“ eingeschlossen und erhalten eine Bezeichnung über die bekannte AS-Klausel. Die APPEND-Methode schließt das RELATE ein wird jedoch mit normalen Klammern „(…)“ begrenzt. Die Verbindung der Spalten des Recordsets der ersten Ebene und der zweiten Ebene werden nicht wie normalerweise gewohnt mit „=“ verbunden sondern mit dem Schlüsselwort „TO“ verbunden und die Namen der Spalten werden als Strings über Hochkommas geschrieben.

Hierarchien in einer einzelnen Tabelle

Jeder der schon öfter mit SQL gearbeitet hat kennt das leidige Problem wenn man versucht in einer Tabelle eine Aggregatfunktion (COUNT, MAX, MIN, …) auszuführen jedoch eigentlich auch die Detaildaten mit anzeigen möchte. Hier entstanden bislang riesige JOINs oder IN-Statements. Auch dies lässt sich über hierarchische Recordsets sehr schön lösen in dem man noch zwei neue Begriffe, nämlich CUMPUTE und BY, in die Syntax mit aufnimmt.

Zum Beispiel möchte man die Daten der Bestelltabelle sehen und dazu die Menge aller verkauften Artikel gruppiert nach Artikelnummern:

SHAPE {SELECT * FROM tbl_Bestellungen} AS Grund

COMPUTE Grund, SUM(Grund.Menge) AS Gesamtmenge

BY Artikelnummer

Das Ergebnis kann dann so aussehen:

Über mehrere Tabellen mit Aggregatfunktionen

Jetzt wird es noch mal etwas komplexer. Bei Relationalen Datenbanken wird in einer Detailtabelle schließlich nur die ID eines Hauptwertes gespeichert (siehe oben die Artikelnummer) und mit dieser können User nur selten wirklich etwas anfangen. Somit müssen wir erst die Daten über ein JOIN miteinander verbinden und darauf dann unser COMPUTE setzen.

Wir lassen uns nun die Daten aus der Artikeltabelle mit denen aus der Bestelltabelle zusammen anzeigen und zusätzlich über die COUNT-Funktion die Summe aller Bestellungen für die einzelnen Artikel ermitteln:

SHAPE {

SELECT 

  A_Name AS Artikel,

  A_Preis AS Preis,

  B_Id AS Bestellnummer,

  B_Menge AS Menge

FROM tbl_Artikel

  INNER JOIN tbl_Bestellungen

  ON tbl_Artikel.A_Id = tbl_Bestellungen.A_Id

} AS AtikelCompute

COMPUTE AtikelCompute, 

        COUNT(AtikelCompute.'Artikel') AS Anzahl

BY 'Artikel'

Das Ergebnis könnte dann so aussehen:

Arbeiten mit hierarchischen Recordsets

Nun haben wir gesehen wie man hierarchische Recordsets aufbaut, noch nicht gesehen wie man diese denn im Code behandelt. Zu Grunde legen wir den nachfolgenden Beispielen das SQL-Statement aus dem Grundbeispiel:

SHAPE {SELECT * FROM tbl_Artikel ORDER BY A_Name} AS ArtikelHierarch

APPEND ({SELECT * FROM tbl_Bestellungen} AS BestellSub

RELATE 'A_ID' TO 'A_ID') AS BestellSub

Wenn man bei einem solchen Recordset die Felder anzeigen lässt wird man seltsamer Weise nur die Felder das Hauptrecordsets sehen und den Namen des Unterrecordsets („BestellSub“), darum wird nun dem DataTypeEnum aus den Feld-Typen des Recordsets ein neuer Wert hinzugefügt. Der adChapter, er liegt vor wenn wir einen Unterdatensatz in einem SHAPE haben. Hierbei handelt es sich eigentlich um ein eigenständiges Recordset. Dieses kann an ein neues Recordset-Objekt übergeben werden und genau wie das erste auch gehandelt werden.

Als Beispiel lassen wir die Detaildaten der Bestellungen für den ersten Artikel in einem DataGrid anzeigen:

Option Explicit

Dim CnT As ADODB.Connection

Dim RsT As ADODB.Recordset

Dim RsDetail As ADODB.Recordset

Private Sub Form_Load()

' SQL-Statement für hierarchisches Recordset

Const sSQL As String = _

    "SHAPE {SELECT * FROM tbl_Artikel ORDER BY A_Name} " & _

    "AS ArtikelHierarch " & _

    "APPEND ({SELECT * FROM tbl_Bestellungen} " & _

    "AS BestellSub " & _

    "RELATE 'A_ID' TO 'A_ID') AS BestellSub"

Dim i As Integer

Set CnT = New ADODB.Connection

With CnT

    .CursorLocation = adUseClient

    .Provider = "MSDataShape"

    .Properties("Data Provider") = "SQLOLEDB.1"

    .Properties("Data Source") = "FLO_MAIN"

    .Properties("Initial Catalog") = "db_Test"

    .Properties("User Id") = "florianr"

    .Properties("Password") = "flo"

    .Open

End With

Set RsT = New ADODB.Recordset

With RsT

    .ActiveConnection = CnT

    .CursorLocation = adUseClient

    .CursorType = adOpenKeyset

    .LockType = adLockOptimistic

    .Source = sSQL

    .Open

End With

' Hierarchisches Recordset an MSHFlexGrid

Set Me.MSHFlexGrid1.DataSource = RsT

For i = 0 To RsT.Fields.Count - 1

    If RsT.Fields(i).Type = adChapter Then

        Set RsDetail = RsT.Fields(i).Value

        Exit For

    End If

Next i

If Not RsDetail Is Nothing Then

    Set Me.DataGrid1.DataSource = RsDetail

End If

End Sub

Nun ist das Recordset RsDetail fest (!) an die Detaildaten angebunden. Wenn man nun durch das Hauptrecordset navigiert werden automatisch die Daten des Unterrecordsets geupdated und an das Detaildatenrecordset übergeben.

Asynchrone Aktionen

Was hat asynchron wieder mit ADO zu tun? Das bedeutet hier, daß viele Aktionen auf einer Datenbank ausgeführt werden müssen laufen können ohne, daß die Anwendung im Vordergrund darauf warten muß. Hierbei wird eine Aktion an den Server weitergegeben und das Programm läuft sofort weiter. Das kann sowohl dafür nützlich sein um die Anwendung bereits für weitere Arbeiten welche nicht direkt mit Arbeiten auf dem Server zu tun haben frei zu geben als auch um beispielsweise über eine Statusbar eine Prozessbar oder sonstiges die Arbeit dem Anwender zu verdeutlichen.

Bei dieser Methode muß jedoch einiges beachtet werden. Um eine asynchrone Aktion abzusetzen muß beim ausführen der Aktion der jeweilige Parameter für die asynchrone Aktion als Option übergeben werden.

Asynchroner Verbindungsaufbau

Bei vielen Datenbankprogrammen ist das leidige Thema bekannt, daß man sich anmeldet und dann erst einige Zeit warten muß bis man auf der Startmaske des Programms landet auf welcher meist noch gar keine Daten angezeigt werden. Mit ADO ist es möglich die Verbindung herzustellen und das Programm bereits weiterlaufen zu lassen. So können z.B. alle anderen rechen- und zeitintensiven Arbeiten des Programms, die ebenfalls am Anfang ausgeführt werden müssen bereits ausgeführt werden. Hierbei muß jedoch beachtet werden, daß Daten erst abgefragt werden können wenn die Verbindung komplett hergestellt worden ist. Die Option die hier übergeben werden muß ist adAsyncConnect.

Hier ein kleines Beispiel das zeigt, daß die Connection begonnen wird, dann der Code weiterläuft und erst wenn die Verbindung steht dies signalisiert wird. Zu diesem Zweck wird einfach vor dem Anfang des Verbindungsaufbaus, nach dem starten und beim Vollenden des Verbindungsaufbaus eine kleine Meldung auf das Form geschrieben:

Option Explicit

Dim WithEvents CnT As ADODB.Connection

Private Sub CnT_ConnectComplete( _

            ByVal pError As ADODB.Error, _

            adStatus As ADODB.EventStatusEnum, _

            ByVal pConnection As ADODB.Connection)

    If adStatus = adStatusOK Then

        Me.Print "Connection steht"

    End If

End Sub

Private Sub Form_Load()

Me.AutoRedraw = True

Me.Print "Anfang connecten"

Set CnT = New ADODB.Connection

With CnT

    .CursorLocation = adUseClient

    .Provider = "MSDataShape"

    .Properties("Data Provider") = "SQLOLEDB.1"

    .Properties("Data Source") = "FLO_MAIN"

    ' NT-Authentifikation

    .Properties("Initial Catalog") = "db_Test"

    .Properties("Integrated Security") = "SSPI"

    .Open Options:=adAsyncConnect

End With

Me.Print "Am connecten..."

End Sub

Wichtig ist auch, daß bei dem Event ConnectComplete überprüft wird ob der Verbindungsaufbau auch erfolgreich war, sonst kann entweder noch mal mit einem höherem CommandTimeout ein versucht werden oder eine Fehlermeldung ausgegeben werden.

Während die Connection noch hergestellt wird kann kein Recordset oder anderes Objekt an diese angebunden werden und kein Execute ausgeführt werden.

Asynchrones Recordset

Hier können asynchrone Abläufe vor allem bei sehr großen Datenmengen sehr hilfreich sein. Wenn man z.B. sehr viele Daten über das Netzwerk transportieren will kann das manchmal sehr lange dauern. Über eine asynchrone Abfrage können jedoch bereits die ersten Daten angezeigt und auch navigiert werden während sich das Recordset noch komplettisiert. Als Option muß hierfür adAsyncFetch übergeben werden.

Hier ebenfalls noch ein Beispiel das zeigt, daß ein Programm bereits weiterlaufen kann und auch durch die Daten navigiert werden können bevor die Abfrage komplett ist. Die hier eingebundene Tabelle tblMany enthält 100.000 Datensätze:

Option Explicit

Dim CnT As ADODB.Connection

Dim WithEvents RsT As ADODB.Recordset

Private Sub Form_Load()

Me.AutoRedraw = True

Me.Print "Anfang connecten"

Set CnT = New ADODB.Connection

With CnT

    .CursorLocation = adUseClient

    .Provider = "MSDataShape"

    .Properties("Data Provider") = "SQLOLEDB.1"

    .Properties("Data Source") = "FLO_MAIN"

    .Properties("Initial Catalog") = "db_Test"

    .Properties("Integrated Security") = "SSPI"

    .Open

End With

Set RsT = New ADODB.Recordset

With RsT

    .ActiveConnection = CnT

    .CursorLocation = adUseClient

    .CursorType = adOpenKeyset

    .LockType = adLockOptimistic

    .Source = "SELECT * FROM tblMany"

    .Open Options:=adAsyncFetch

End With

RsT.MoveNext

Me.Print "Am Datan auslesen..."

Set Me.DataGrid1.DataSource = RsT

End Sub

Private Sub RsT_FetchComplete( _

            ByVal pError As ADODB.Error, _

            adStatus As ADODB.EventStatusEnum, _

            ByVal pRecordset As ADODB.Recordset)

    Me.Print "Recordset geladen"

End Sub

Auch hier wird die Aktion das die Daten vollständig sind durch eine Methode des Recordsets angezeigt und die verschiedenen Stati der Einfachheit halber auf die Form geschrieben.

Asyncrone Commands

Gerade hier kann man mit asynchronen Arbeiten eine Anwendung wesentlich optimieren. StoredProcedures können manchmal einige Zeit in Anspruch nehmen und das kann dann wie beim Recordset verbessert werden. Hierfür ist die zu übergebende Option adAsyncExecute. Dieser Parameter kann auch für eine Connection verwendet werden um ein Execute zu optimieren.

Wenn von der SQL-Prozedur keine Daten zurückgeliefert werden (z.B. bei Löschungen) kann dies ebenfalls als Option mit adExecuteNoRecords angegeben werden welche die Anwendung noch mal optimiert, so kann eine StoredProcedure ohne weiteres 30 Sekunden auf dem Server dauern ohne das dies die Anwendung stört.

Hier ein Beispiel für optimiertes ausführen einer StoredProcedure ohne Datenrückgabe:

Option Explicit

Dim CnT As ADODB.Connection

Private Sub Form_Load()

Dim Cmd As ADODB.Command

Set CnT = New ADODB.Connection

With CnT

    .CursorLocation = adUseClient

    .Provider = "MSDataShape"

    .Properties("Data Provider") = "SQLOLEDB.1"

    .Properties("Data Source") = "FLO_MAIN"

    .Properties("Initial Catalog") = "db_Test"

    .Properties("Integrated Security") = "SSPI"

    .Open

End With

Set Cmd = New ADODB.Command

With Cmd

    .ActiveConnection = CnT

    .CommandType = adCmdStoredProc

    .CommandText = "sp_Delete_Bestell"

    .Parameters.Refresh

    .Parameters("@BestellId") = 123

    .Execute Options:=adAsyncExecute + adExecuteNoRecords

End With

End Sub

Sicherheit bei asynchronen Aktionen

Eines muß bei asynchronen Aktionen beachtet werden, eine Connection kann zwar mehrere Commands und Recordsets gleichzeitig verwalten, jedoch nicht mehrere Aktionen gleichzeitig ausführen. Hierzu können Sie sich aber sehr leicht über eine kleine Funktion helfen welche den Status der Connection abfragt und solange wartet. Dies lässt sich am besten mit der Sleep-API oder über ein DoEvents (hier CheckForExec) lösen:

Option Explicit

Private Declare Sub Sleep Lib "kernel32" (ByVal dwMilliseconds As Long)

Private Function CheckForExec( _

            Conn As ADODB.Connection, _

            Optional TimeOut As Long = 15 _

        ) As Boolean

Dim dTemp As Date

If Conn.State = adStateClosed Then  ' Wenn Verbindung geschlossen,

    Conn.Open , , , adOpenAsync     ' dann öffnen

End If

' Wenn Status nicht OK, dann Schleife beginnen

If Conn.State = adStateExecuting Or Conn.State = adStateFetching Then

    dTemp = Now

    Do

        Sleep 100   ' 100 Millisekunden warten

        ' Nach vorgegebener Max-Zeit beenden

        If DateDiff("s", dTemp, Now) >= TimeOut Then

            MsgBox "Die Verbinung zum Server ist gerade ausgelastet. " & _

                "Bitte warten Sie etwas und führen Sie dann nochmal aus."

            CheckForExec = False

            Exit Function

        End If

    Loop Until Conn.State = adStateOpen

End If

CheckForExec = True

End Function

Private Sub Command1_Click()

Set RsT = New ADODB.Recordset

    With RsT

        .ActiveConnection = CnT

        .LockType = adLockOptimistic

        .CursorLocation = adUseClient

        .CursorType = adOpenKeyset

        .Source = "SELECT * FROM Grunddaten"

        If Not CheckForExec(CnT, 15) Then

            Exit Sub

        End If

        .Open

    End With

End Sub

Schlusswort

Obwohl hier doch recht ausführlich auf ADO eingegangen wurde gibt es noch einige Methoden und Eigenschaften die ausgelassen wurden. Sie haben gesehen, daß es sich beim Thema Datenbanken um ein sehr komplexes Thema handelt und mit dem hier beschriebenen ist eigentlich nur ein Grundstock gesetzt. Die eigentliche Komplexität besteht erst darin das Gerüst einer kompletten relationalen Datenbank zu bilden und die vielen Möglichkeiten der Datenbankprogrammierung richtig einzusetzen.

Nicht erwähnt

Wer hier die Thematik ADOX und SQLDMO vielleicht vermisst hat, jedes dieser Themen ist in sich wieder so komplex, daß es ohne weiteres in ein eigenes Tutorial gehört.

Das seit ADO 2.5 neue Record-Objekt konnte aus bislang mangelnder Erfahrung damit nicht beschrieben werden. Ebenfalls das Stream-Objekt ist nur sehr kurz angesprochen. Die eigentliche Stärke dieser Objekte wird eventuell noch nachgeliefert.

